
Changelog
    


  
    
  
  v0.10.0 (2025-06-17)



  
    
  
  Enhancements


	[Nx] Improvements to documentation, including new advanced guides and a revamped Getting Started section and cheatsheets for comparison between Nx and NumPy
	[Nx.Defn.Expr] floating-point constants now take on the precision of the surrounding operations, improving precision of f64 calculations
	[Nx] Nx.to_pointer/2 and Nx.from_pointer/5 now raise on errors
	[Nx.LinAlg] Nx.LinAlg.lu/2 now has a defn implementation
	[Nx.Defn.Graph] Create new module for splitting an Nx.Defn.Expr computation into connected computation stages
	[Nx.Testing] New module for testing functions that rely on Nx
	[Nx.Defn.Evaluator] Add new debugging option for saving all intermediate results to disk or printing to stdout


  
    
  
  v0.9.2 (2024-11-16)



  
    
  
  Bug fixes


	[Nx] Fix deprecation warnings on latest Elixir
	[Nx.LinAlg] Fix least_squares implementation
	[Nx.Random] Fix Nx.Random.shuffle repeating a single value in certain cases on GPU


  
    
  
  v0.9.1 (2024-10-08)



  
    
  
  Deprecations


	[Nx] Deprecate Nx.Defn.stream


  
    
  
  v0.9.0 (2024-09-26)



  
    
  
  Enhancements


	[Nx] Add 8-bit Floating Point numerical type
	[Nx] Add quantized int types (s2, s4, u2, u4)


  
    
  
  Bug fixes


	[Nx.LinAlg] Minor range slicing fixes on QR decomposition
	[Nx] Nx.Defn.Grad now supports more vectorization cases


  
    
  
  Deprecations and incompatibilities


	[Nx] Default integer type is now s32
	[Nx] Interface breaking changes for Nx.to_pointer and Nx.from_pointer


  
    
  
  v0.8.0 (2024-08-19)



  
    
  
  Enhancements


	[Nx] Add Nx.to_pointer/2 and Nx.from_pointer/5
	[Nx] Introduce ~VEC sigil for 1d tensors
	[Nx] Introduce ~MAT sigil for 2d tensors
	[Nx] Implement stack as a callback for performance
	[Nx] Make take an optional callback
	[Nx] Make take_along_axis an optional callback
	[Nx.LinAlg] Support :keep_axes in eigh


  
    
  
  Bug fixes


	[Nx] Fix a bug with gather when indices had more dimensions than the input tensor
	[Nx] Fix min/max value for 16 bit signed type
	[Nx] Fix argmax/argmin behaviour with NaNs
	[Nx.Serving] Fix a bug where streaming responses were never closing


  
    
  
  Deprecations and incompatibilities


	[Nx] Deprecate ~V in favor of ~VEC
	[Nx] Deprecate ~M in favor of ~MAT
	[Nx] Remove Nx.map/2


  
    
  
  v0.7.1 (2024-02-27)


	[Nx.LinAlg] Minor speed up to Nx.LinAlg.qr/2 default implementation


  
    
  
  v0.7.0 (2024-02-22)



  
    
  
  Enhancements


	[Nx] Add Nx.fft2 and Nx.ifft2
	[Nx] Add Nx.fill/2
	[Nx] Implement QR decomposition as optional callback
	[Nx] Support :type option in argmin/argmax
	[Nx] Default all sorting operations to unstable sorting (pass stable: true to change it)
	[Nx.BinaryBackend] Improve performance of Nx.concatenate/2
	[Nx.Defn] Support a mapping function in print_value/2
	[Nx.Defn] Add Nx.Defn.Compiler.__to_backend__/1 callback
	[Nx.LinAlg] Add Nx.least_squares/2


  
    
  
  Bug fixes


	[Nx.Constants] Fix min and max finite values for :bf16
	[Nx.Defn] Do not discard arguments on optional grads


  
    
  
  Incompatible changes


	[Nx] Default to non-stable sorting
	[Nx] Remove deprecated random_uniform, random_normal, shuffle
	[Nx.Defn] Nx.Defn.rewrite_types/2 has been removed


  
    
  
  v0.6.4 (2023-11-13)



  
    
  
  Enhancements


	[Nx] Allow non-scalar tensors on access


  
    
  
  Bug fixes


	[Nx] Improve the :axes option in gather, indexed_add, and indexed_put
	[Nx] Fix grad of gather, indexed_add, and indexed_put with axes
	[Nx.BinaryBackend] Fix sorting of negative infinity
	[Nx.BinaryBackend] Always sort NaN last
	[Nx.Serving] Fix Nx.Batch padding with multi-device backends


  
    
  
  v0.6.3 (2023-11-09)



  
    
  
  Enhancements


	[Nx] Allow non-scalars as updates on indexed_add and indexed_put
	[Nx] Allow non-scalars as return of gather
	[Nx] Support the :axes option in gather, indexed_add, and indexed_put
	[Nx] Add Nx.covariance
	[Nx] Support :type in argsort
	[Nx] Support :stable option in argsort for future compatibility
	[Nx.Serving] Add :weight option for static load balancing


  
    
  
  Bug fixes


	[Nx] Cast input types on slicing
	[Nx.Defn] Support vectorized tensors in grad
	[Nx.Defn] Fix bugs when diffing tensor expressions
	[Nx.Serving] Handle serving getting stuck on timer messages


  
    
  
  v0.6.2 (2023-09-21)



  
    
  
  Enhancements


	[Nx.Serving] Add Nx.Serving.batch_size/2 and perform batch splitting on run
	[Nx.Serving] Support input streaming


  
    
  
  v0.6.1 (2023-09-12)



  
    
  
  Enhancements


	[Nx] Add multivariate normal distribution
	[Nx.Serving] Automatically split exceeding batch sizes


  
    
  
  Bug fixes


	[Nx] Fix Nx.pad/2 with different backends
	[Nx] Fix Nx.clip/3 with non-finite values
	[Nx.Serving] Emit batches as they arrive in Nx.Serving.streaming/2
	[Nx.Serving] Ensure batch key is preserved when a batch is split


  
    
  
  v0.6.0 (2023-08-15)



  
    
  
  Enhancements


	[Nx] Add constant creation helpers such as u8, f32, etc
	[Nx] Implement Bluestein's algorithm for fft and ifft in the binary backend
	[Nx] Support range with steps when accessing tensors
	[Nx] Support vectorization via Nx.vectorize/2, Nx.devectorize/2, Nx.revectorize/2, Nx.reshape_vectors/2, and Nx.broadcast_vectors/2
	[Nx] Add Nx.logsumexp/2
	[Nx] Add Nx.split/3
	[Nx] Add Nx.tri/2, Nx.triu/2, Nx.tril/2
	[Nx] Introduce a new serialization format that is more suitable to memory mapping
	[Nx.Defn] Consider Inspect.Opts limit when pretty printing Nx.Defn expressions
	[Nx.Serving] Support multiple batch keys in Nx.Serving
	[Nx.Serving] Support streaming in Nx.Serving


  
    
  
  Bug fixes


	[Nx] Fix from_numpy with 1-byte width arrays
	[Nx] Fix cases where pretty printing large Nx.Defn expressions would take a long time
	[Nx] Fix reduce_min/reduce_max for non-finite values


  
    
  
  Deprecations


	[Nx.Serving] The post-processing function must now be a two-arity function that receives the {output, metadata} as a pair or the stream


  
    
  
  Breaking changes


	[Nx.Serving] The nx.serving.postprocessing telemetry event no longer receives the serving output or serving metadata as event metadata


  
    
  
  v0.5.3 (2023-04-14)



  
    
  
  Bug fixes


	[Nx.Defn] Fix compilation error when Elixir compiler has column tracking enabled
	[Nx.LinAlg] Fix cases where determinant could return NaN
	[Nx.LinAlg] Fix SVD when working with f16 and bf16


  
    
  
  v0.5.2 (2023-03-21)



  
    
  
  Enhancements


	[Nx.Random] Add stop_grad to Nx.Random creation functions
	[Nx.Serving] Reduce references sent through serving


  
    
  
  Bug fixes


	[Nx] Fix Nx.mode with :axis option


  
    
  
  v0.5.1 (2023-02-18)


Require Elixir v1.14.

  
    
  
  Enhancements


	[Nx] Support any container or lazy container in stack/concatenate
	[Nx] Add Nx.top_k/2
	[Nx] Add Nx.to_list/1
	[Nx] Improve shape validation in Nx.concatenate/2
	[Nx.Constants] Add pi, e, and euler_gamma
	[Nx.Random] Raise if a non-unary rank tensor is given as probabilities to Nx.Random.choice/4
	[Nx.Random] Make samples optional in Nx.Random.choice/3


  
    
  
  v0.5.0 (2023-02-10)



  
    
  
  Enhancements


	[Nx] Support serialization of containers
	[Nx] Rename Nx.power to Nx.pow
	[Nx] Add Nx.reflect and Nx.linspace
	[Nx.Defn] Raise at compile time for invalid defn if/cond usage
	[Nx.LinAlg] Support full_matrices? in SVD
	[Nx.LinAlg] Add Nx.LinAlg.matrix_rank
	[Nx.Random] Add Nx.Random.choice and Nx.Random.shuffle
	[Nx.Serving] Add distributed² serving by distributing over devices (GPUs/CPUs) as well as nodes
	[Nx.Serving] Add telemetry to Nx.Serving callbacks


  
    
  
  Backwards incompatible changes


	[Nx] from_numpy and from_numpy_archive have been replaced by load_numpy! and load_numpy_archive!
	[Nx.Defn.Evaluator] Do not force GC on evaluator


  
    
  
  v0.4.2 (2023-01-13)



  
    
  
  Enhancements


	[Nx] Allow tensors to be given on Nx.tensor/2
	[Nx] Add Nx.with_default_backend/2
	[Nx] Add :axes option to Nx.flatten/2
	[Nx] Add :axes option to Nx.weighted_mean/2
	[Nx.Defn] Warn if Nx.tensor/2 first-argument is not constant inside defn
	[Nx.LinAlg] Add Nx.LinAlg.pinv/1
	[Nx.LinAlg] Optimize and handle more cases in Nx.LinAlg.svd/1


  
    
  
  Bug fixes


	[Nx] Respect fortran order in loading from numpy
	[Nx.Defn] Render containers in compile error type+shape mismatch
	[Nx.Defn] Restore pdict state after compilation


  
    
  
  v0.4.1 (2022-12-07)



  
    
  
  Enhancements


	[Nx] Add Nx.Batch and Nx.Serving
	[Nx] Implement Nx.Container for numbers, complex, and tensors for completeness
	[Nx] Support batches in Nx.eye/2


  
    
  
  Bug fixes


	[Nx] Keep input tensor names on associative scan
	[Nx.BinaryBackend] Differentiate between complex and real output in as_type
	[Nx.BinaryBackend] Fix loss of precision in Nx.complex/2
	[Nx.BinaryBackend] Preserve NaNs in window and reduce operations
	[Nx.Random] Do not return infinity on normal/2 for f16


  
    
  
  v0.4.0 (2022-10-25)



  
    
  
  Enhancements


	[Nx] Add Nx.rename/2, Nx.median/2, Nx.weighted_mean/3, and Nx.mode/2
	[Nx] Implement cumulative operations using associative scan for improved performance
	[Nx.Constants] Add min and max
	[Nx.Defn] Allow lists and functions anywhere as arguments in defn, jit and compile
	[Nx.Defn] Add Nx.LazyContainer that allows a data-structure to lazily define tensors
	[Nx.Defn] Allow tensors and ranges as generators inside while
	[Nx.Defn] Add debug_expr/2 and debug_expr_apply/3
	[Nx.Defn.Evaluator] Calculate cache lifetime to reduce memory usage on large numerical programs
	[Nx.LinAlg] Handle Hermitian matrices in eigh
	[Nx.LinAlg] Support batched operations in adjoint, cholesky, determinant, eigh, invert, lu, matrix_power, solve, svd, and triangular_solve
	[Nx.Random] Support pseudo random number generators algorithms


  
    
  
  Bug fixes


	[Nx] Perform window_reduce/reduce operations from infinity and negative infinity
	[Nx.Defn] Ensure defnp emits warnings when unused
	[Nx.Defn] Warn on unused variables in while


  
    
  
  Deprecations


	[Nx] Deprecate tensor as shape in Nx.eye/2 and Nx.iota/2
	[Nx] Deprecate Nx.random_uniform/2 and Nx.random_normal/2


  
    
  
  v0.3.0 (2022-08-13)



  
    
  
  Enhancements


	[Nx] Improve support for non-finite values in Nx.broadcast/2, Nx.all_close/2, and more
	[Nx] Add Nx.is_nan/1 and Nx.is_infinite/1
	[Nx] Support booleans in Nx.tensor/2
	[Nx] Add Nx.fft/2 and Nx.ifft/2
	[Nx] Rename Nx.logistic/1 to Nx.sigmoid/1
	[Nx] Add Nx.put_diagonal/3 and Nx.indexed_put/3
	[Nx] Add :reverse to cummulative functions
	[Nx] Add Nx.to_batched/3 which returns a stream
	[Nx] Support batched tensors in Nx.LinAlg.qr/1
	[Nx.Defn] Add Nx.Defn.compile/3 for precompiling expressions
	[Nx.Defn] Add deftransform/2 and deftransformp/2 for easier to define transforms
	[Nx.Defn] Add div/2
	[Nx.Defn] Support case/2, raise/1, and raise/2
	[Nx.Defn] Support booleans in if, cond, and boolean operators
	[Nx.Defn] Perform branch elimitation in if and cond and execute branches lazily
	[Nx.Defn.Evaluator] Garbage collect after evaluation (it can be disabled by setting the :garbage_collect compiler option to false)


  
    
  
  Deprecations


	[Nx] Nx.to_batched_list/3 is deprecated in favor of Nx.to_batched/3
	[Nx.Defn] transform/2 is deprecated in favor of deftransform/2 and deftransformp/2
	[Nx.Defn] assert_shape/2 and assert_shape_pattern/2 are deprecated in favor of case/2 + raise/2
	[Nx.Defn] inspect_expr/1 and inspect_value/1 are deprecated in favor of print_expr/1 and print_value/1 respectively


  
    
  
  v0.2.1 (2022-06-04)



  
    
  
  Enhancements


	[Nx] Improve support for non-finite values in Nx.tensor/1
	[Nx] Use iovec on serialization to avoid copying binaries
	[Nx.BinaryBackend] Improve for complex numbers in Nx.tensor/1
	[Nx.Defn] Improve for complex numbers inside defn


  
    
  
  Bug fixes


	[Nx] Properly normalize type in Nx.from_binary/3
	[Nx.Defn] Raise on Nx.Defn.Expr as JIT argument
	[Nx.Defn.Evaluator] Handle concatenate arguments on evaluator


  
    
  
  v0.2.0 (2022-04-28)


This version requires Elixir v1.13+.

  
    
  
  Enhancements


	[Nx] Support atom notation as the type option throughout the API (for example, :u8, :f64, etc)
	[Nx] Add support for complex numbers (c64, c128)
	[Nx] Add Nx.cumulative_sum/2, Nx.cumulative_product/2, Nx.cumulative_min/2, Nx.cumulative_max/2
	[Nx] Add Nx.conjugate/1, Nx.phase/1, Nx.real/1, and Nx.imag/1
	[Nx] Initial support for NaN and Infinity
	[Nx] Add :axis option to Nx.shuffle/2
	[Nx] Add Nx.axis_index/2
	[Nx] Add Nx.variance/2 to Nx.standard_deviation/2
	[Nx] Rename Nx.slice_axis/3 to Nx.slice_along_axis/4
	[Nx.Backend] Add support for optional backends
	[Nx.Constants] Provide a convenient module to host constants
	[Nx.Defn] Improve error messages throughout the compiler


  
    
  
  v0.1.0 (2022-01-06)


First release.


  

    
What is Nx?
    

Nx is a numerical computing library in Elixir. Since Elixir's primary numerical data types and structures are not optimized for numerical programming, Nx is the fundamental package built to bridge this gap.
Elixir Nx smoothly integrates typed, multidimensional data called tensors).
Nx has four primary capabilities:
	Tensors hold typed data in multiple, optionally named dimensions.
	Numerical definitions, known as defn, support custom code with
tensor-aware operators and functions.
	Automatic differentiation, also known as
autodiff, supports common computational scenarios
such as machine learning, simulations, curve fitting, and probabilistic models.
	Broadcasting, which is a term for element-by-element operations. Most of the Nx operations
make use of automatic implicit broadcasting. You can see more on broadcasting
here.

Nx tensors can hold unsigned integers (u2, u4, u8, u16, u32, u64),
signed integers (s2, s4, s8, s16, s32, s64),
floats (f8, f16, f32, f64), brain floats (bf16), and complex (c64, c128).
Tensors support backends implemented outside of Elixir, such as Google's
Accelerated Linear Algebra (XLA) and PyTorch.
Numerical definitions provide compiler support to allow just-in-time compilation
targetting specialized processors to speed up numeric computation including
TPUs and GPUs.

  
    
  
  What are Tensors?


In Nx, we express multi-dimensional data using typed tensors. Simply put,
a tensor is a multi-dimensional array with a predetermined shape and
type. To interact with them, Nx relies on tensor-aware operators rather
than Enum.map/2 and Enum.reduce/3.
It allows us to work with the central theme in numerical computing, systems of equations,
which are often expressed and solved with multidimensional arrays.
For example, this is a two dimensional array:
$$
\begin{bmatrix}
  1 & 2 \\\\
  3 & 4
\end{bmatrix}
$$
As elixir programmers, we can typically express a similar data structure using a list of lists,
like this:
[
  [1, 2],
  [3, 4]
]
This data structure works fine within many functional programming
algorithms, but breaks down with deep nesting and random access.
On top of that, Elixir numeric types lack optimization for many numerical
applications. They work fine when programs
need hundreds or even thousands of calculations. However, they tend to break
down with traditional STEM applications when a typical problem
needs millions of calculations.
To solve for this, we can simply use Nx tensors, for example:
Nx.tensor([[1,2],[3,4]])

Output:
#Nx.Tensor<
s32[2][2]
[
  [1, 2],
  [3, 4]
]
To learn Nx, we'll get to know tensors first. The following overview will touch
on the major features. The advanced section of the documentation will take a deep dive into working
with tensors in detail, automatic differentiation, and backends.


  

    
Installation
    

The only prerequisite for installing Nx is Elixir itself. If you don't have Elixir installed
in your machine you can visit this installation page.
There are several ways to install Nx (Numerical Elixir), depending on your project type and needs.

  
    
  
  Using Mix in a standardElixir Project


If you are working inside a Mix project, the recommended way to install Nx is by adding it to your mix.exs dependencies:
	Open mix.exs and modify the deps function:

defp deps do
  [
    {:nx, "~> 0.9"}  # Install the latest stable version
  ]
end
	Fetch the dependencies, run on the terminal:

mix deps.get


  
    
  
  Installing Nx from GitHub (Latest Development Version)


If you need the latest, unreleased features, install Nx directly from the GitHub repository.
	Modify mix.exs:

defp deps do
  [
    {:nx, github: "elixir-nx/nx", branch: "main", sparse: "nx"}
  ]
end
	Fetch dependencies:

mix deps.get


  
    
  
  Installing Nx in a Standalone Script (Without a Mix Project)


If you don’t have a Mix project and just want to run a standalone script, use Mix.install/1 to dynamically fetch and install Nx.
Mix.install([:nx])

require Nx

tensor = Nx.tensor([1, 2, 3])
IO.inspect(tensor)
Run the script with:
elixir my_script.exs

Best for: Quick experiments, small scripts, or one-off computations.

  
    
  
  Installing the Latest Nx from GitHub in a Standalone Script


To use the latest development version in a script (without a Mix project):
Mix.install([
  {:nx, github: "elixir-nx/nx", branch: "main", sparse: "nx"}
])

require Nx

tensor = Nx.tensor([1, 2, 3])
IO.inspect(tensor)
Run:
elixir my_script.exs

Best for: Trying new features from Nx without creating a full project.

  
    
  
  Installing Nx with EXLA for GPU Acceleration


To enable GPU/TPU acceleration with Google’s XLA backend, install Nx along with EXLA:
	Modify mix.exs:

defp deps do
  [
    {:nx, "~> 0.9"},
    {:exla, "~> 0.9"}  # EXLA (Google XLA Backend)
  ]
end
	Fetch dependencies:

mix deps.get

	Run with EXLA enabled:

Nx.default_backend(EXLA.Backend)
Nx.Defn.default_options(compiler: EXLA)
Best for: Running Nx on GPUs or TPUs using Google’s XLA compiler.

  
    
  
  Installing Nx with Torchx for PyTorch Acceleration


To run Nx operations on PyTorch’s backend (LibTorch):
	Modify mix.exs:

defp deps do
  [
    {:nx, "~> 0.9"},
    {:torchx, "~> 0.9"}  # PyTorch Backend
  ]
end

	Fetch dependencies:

mix deps.get

	Run with Torchx enabled:

Nx.default_backend(Torchx.Backend)
Best for: Deep learning applications with PyTorch acceleration.


  

    
Nx Quickstart
    


  
    
  
  Prerequisites


To properly use Nx, you will need to know a bit of Elixir. For a refresher, check out the
Elixir Getting Started Guide.
To work on the examples you can run using the "Run in Livebook" button in this page.
Learning Objectives
This is a overview of Nx tensors. In this section, we'll look at some of the various tools for
creating and interacting with tensors.
After reading, you should be able to understand:
	Create 1, 2 and N-dimensional tensors in Nx;
	How to index, slice and iterate through tensors;
	Basic tensor functions;
	How to apply some linear algebra operations to n-dimensional tensors without using for-loops;
	Axis and shape properties for n-dimensional tensors.


  
    
  
  The Basics


First, let's install Nx with Mix.install.
Mix.install([
  {:nx, "~> 0.9"}
])
The IEx.Helpers module will assist our exploration of core tensor concepts.
import IEx.Helpers

  
    
  
  Creating tensors


The argument for Nx.tensor/1 must be one of:
	a tensor;
	a number (which means the tensor is scalar/zero-dimensional);
	a boolean (also scalar/zero-dimensional);
	an arbitrarily nested list of numbers and booleans
	the special atoms :nan, :infinity, :neg_infinity, which represent non-finite numbers which are not supported by Elixir floats.

If a new tensor is allocated, it will be allocated in the backend defined by the :backend option.
If it is not provided, Nx.default_backend/0 will be used instead.
Examples
A number returns a tensor of zero dimensions, also known as a scalar:
Nx.tensor(0)
Nx.tensor(1.0)
A list returns a one-dimensional tensor, also known as a vector:
Nx.tensor([1, 2, 3])
Nx.tensor([1.2, 2.3, 3.4, 4.5])
Higher dimensional tensors are also possible:
Nx.tensor([[1, 2, 3], [4, 5, 6]])
Nx.tensor([[1, 2], [3, 4], [5, 6]])
Nx.tensor([[[1, 2], [3, 4], [5, 6]], [[-1, -2], [-3, -4], [-5, -6]]])
Tensors can also be given as inputs, which is useful for functions that don't care
about the input kind:
Nx.tensor(Nx.tensor([1, 2, 3]))

  
    
  
  Naming dimensions


You can provide names for tensor dimensions. Names are atoms:
Nx.tensor([[1, 2, 3], [4, 5, 6]], names: [:x, :y])
Names make your code more expressive:
Nx.tensor([[[1, 2, 3], [4, 5, 6], [7, 8, 9]]], names: [:batch, :height, :width])
We created a tensor of the shape {3, 3}, and two axes named height and width.
You can also leave dimension names as nil (which is the default):
Nx.tensor([[[1, 2, 3], [4, 5, 6], [7, 8, 9]]], names: [:batch, nil, nil])
However, you must provide a name for every dimension in the tensor. For example,
the following code snippet raises an error because 1 name is given, but there are 3 dimensions:
Nx.tensor([[[1, 2, 3], [4, 5, 6], [7, 8, 9]]], names: [:batch])

  
    
  
  Indexing and Slicing tensor values


We can get any cell of the tensor:
tensor = Nx.tensor([[1, 2], [3, 4]], names: [:y, :x])
tensor[[0, 1]]
Negative indices will start counting from the end of the axis.
-1 is the last entry, -2 the second to last and so on.
tensor = Nx.tensor([[1, 2], [3, 4], [5, 6]], names: [:y, :x])
tensor[[-1, -1]]
We can also get a whole dimension:
tensor[x: 1]
or a range:
tensor[y: 0..1]
tensor[[.., 1]] will achieve the same result as tensor[x: 1].
This is because Elixir has the syntax sugar .. for a 0..-1//1 range.

  
    
  
  Tensor shape and reshape


Nx.shape(tensor)
We can also create a new tensor with the given shape using Nx.reshape/2:
Nx.reshape(tensor, {1, 4}, names: [:batches, :values])
This operation generally reuses all of the tensor data and simply
changes the metadata, so it has no notable cost.
The new tensor has the same type, but a new shape.

  
    
  
  Floats and Complex numbers


Besides single-precision (32 bits) floats, Nx floating-point numbers can also have other kinds of precision,
such as half-precision (16) or double-precision (64):
Nx.tensor([0.0, 0.2, 0.4, 1.0], type: :f16)
Nx.tensor([0.0, 0.2, 0.4, 1.0], type: :f64)
Brain floats are also supported:
Nx.tensor([0.0, 0.2, 0.4, 1.0], type: :bf16)
Certain backends and compilers support 8-bit floats. The precision
implementation of 8-bit floats may change per backend, so you must be careful
when transferring data across different backends. Nx.BinaryBackend implements F8E5M2:
Nx.tensor([1, 2, 3], type: :f8)
In all cases, the non-finite values negative infinity (-Inf), infinity (Inf),
and "not a number" (NaN) can be represented by the atoms :neg_infinity,
:infinity, and :nan, respectively:
Nx.tensor([:neg_infinity, :nan, :infinity])
Finally, complex numbers are also supported in tensors, in both 32-bit and 64-bit precision:
Nx.tensor(Complex.new(1, -1))
Check out the documentation for Nx.tensor/2 for more documentation on the accepted options.

  
    
  
  Basic operations


Nx supports element-wise arithmetic operations for tensors and broadcasting when necessary.

  
    
  
  Addition


Nx.add/2: Adds corresponding elements of two tensors.
a = Nx.tensor([1, 2, 3])
b = Nx.tensor([0, 1, 2])
Nx.add(a, b)

  
    
  
  Subtraction


Nx.subtract/2: Subtracts the elements of the second tensor from the first.
a = Nx.tensor([10, 20, 30])
b = Nx.tensor([0, 1, 2])
Nx.subtract(a, b)

  
    
  
  Multiplication


Nx.multiply/2: Multiplies corresponding elements of two tensors.
a = Nx.tensor([2, 3, 4])
b = Nx.tensor([0, 1, 2])
Nx.multiply(a, b)

  
    
  
  Division


Nx.divide/2: Divides the elements of the first tensor by the second tensor.
a = Nx.tensor([10, 30, 40])
b = Nx.tensor([5, 6, 8])
Nx.divide(a, b)

  
    
  
  Exponentiation


Nx.pow/2: Raises each element of the first tensor to the power of the corresponding element in the second tensor.
a = Nx.tensor([2, 3, 4])
b = Nx.tensor([2])
Nx.pow(a, b)

  
    
  
  Quotient


Nx.quotient/2: Returns a new tensor where each element is the integer division (div/2).
a = Nx.tensor([10, 20, 30])
b = Nx.tensor([3, 7, 4])

Nx.quotient(a, b)

  
    
  
  Remainder


Nx.remainder/2: Computes the integer division remainder.
a = Nx.tensor([27, 32, 43])
b = Nx.tensor([2, 3, 4])
Nx.remainder(a , b)

  
    
  
  Negation


Nx.negate/1: Negates each element of a tensor.
a = Nx.tensor([2, 3, 4])
Nx.negate(a)

  
    
  
  Square Root


Nx.sqrt/1: Computes the element-wise square root.
a = Nx.tensor([4, 9, 16])
Nx.sqrt(a)

  
    
  
  Element-Wise Comparison


The following operations returns a u8 tensor where 1 represents true and 0 represents false.

  
    
  
  Equality and Inequality


Nx.equal/2, Nx.not_equal/2
a = Nx.tensor([4, 9, 16])
b = Nx.tensor([4, 9, -16])
Nx.equal(a, b)
a = Nx.tensor([4, 9, 16])
b = Nx.tensor([4.0, 9.0, -16.0])
Nx.not_equal(a, b)

  
    
  
  Greater and Less


Nx.greater/2, Nx.less/2
a = Nx.tensor([4, 9, 16])
b = Nx.tensor([4, 8, 17])
Nx.greater(a, b)
a = Nx.tensor([4, 9, 16])
b = Nx.tensor([4.2, 9.0, 15.9])
Nx.less(a, b)

  
    
  
  Greater_Equal and Less_Equal


Nx.greater_equal/2, Nx.less_equal/2
a = Nx.tensor([4, 9, 16])
b = Nx.tensor([4, 8, 17])

Nx.greater_equal(a, b)
a = Nx.tensor([4, 9, 16])
b = Nx.tensor([4.2, 9.0, 15.9])

Nx.less_equal(a, b)

  
    
  
  Aggregate functions


These operations aggregate values across tensor axes.
See also the aggregation guide
for a more in-depth exploration on the subject.

  
    
  
  Sum


Nx.sum/1: Sums all elements.
a = Nx.tensor([[4, 9, 16], [4.2, 9.0, 16.7]])
Nx.sum(a)

  
    
  
  Mean


Nx.mean/1: Computes the mean value of the elements.
a = Nx.tensor([[4, 9, 16], [4.2, 9.0, 16.7]])
Nx.mean(a)

  
    
  
  Product


Nx.product/1: Computes the product of the elements.
a = Nx.tensor([[4, 9, 16], [4.2, 9.0, 16.7]])
Nx.product(a)

  
    
  
  Matrix Multiplication


Nx.dot/4: Computes the generalized dot product between two tensors, operating on specific contracting axes.
t1 = Nx.tensor([[1, 2], [3, 4]], names: [:x, :y])
t2 = Nx.tensor([[10, 20], [30, 40]], names: [:height, :width])
Nx.dot(t1, [0], t2, [0])


  

    
Broadcasting
    

The dimensions of tensors in an operator don't always match.
For example, you might want to subtract a 1 from every
element of a {2, 2}-shaped tensor, like this:
$$
\begin{bmatrix}
  1 & 2 \\\\
  3 & 4
\end{bmatrix} - 1 =
\begin{bmatrix}
  0 & 1 \\\\
  2 & 3
\end{bmatrix}
$$
Mathematically, this is the same as:
$$
\begin{bmatrix}
  1 & 2 \\\\
  3 & 4
\end{bmatrix} -
\begin{bmatrix}
  1 & 1 \\\\
  1 & 1
\end{bmatrix} =
\begin{bmatrix}
  0 & 1 \\\\
  2 & 3
\end{bmatrix}
$$
This means we need a way to convert 1 to a {2, 2}-shaped tensor.
Nx.broadcast/2 solves that problem. This function takes
a tensor or a scalar and a shape.
Mix.install([
  {:nx, "~> 0.9"}
])


Nx.broadcast(1, {2, 2})
This call takes the scalar 1 and translates it
to a compatible shape by copying it. Sometimes, it's easier
to provide a tensor as the second argument, and let broadcast/2
extract its shape:
tensor = Nx.tensor([[1, 2], [3, 4]])
Nx.broadcast(1, tensor)
The code broadcasts 1 to the shape of tensor. In many operators
and functions, the broadcast happens automatically:
Nx.subtract(tensor, 1)
This result is possible because Nx broadcasts both tensors
in subtract/2 to compatible shapes. That means you can provide
scalar values as either argument:
Nx.subtract(10, tensor)
Or subtract a row or column. Mathematically, it would look like this:
$$
\begin{bmatrix}
  1 & 2 \\\\
  3 & 4
\end{bmatrix} -
\begin{bmatrix}
  1 & 2
\end{bmatrix} =
\begin{bmatrix}
  0 & 0 \\\\
  2 & 2
\end{bmatrix}
$$
which is the same as this:
$$
\begin{bmatrix}
  1 & 2 \\\\
  3 & 4
\end{bmatrix} -
\begin{bmatrix}
  1 & 2 \\\\
  1 & 2
\end{bmatrix} =
\begin{bmatrix}
  0 & 0 \\\\
  2 & 2
\end{bmatrix}
$$
This rewrite happens in Nx as well, through a broadcast operation. We want to
broadcast the tensor [1, 2] to match the {2, 2} shape:
Nx.broadcast(Nx.tensor([1, 2]), {2, 2})
The subtract function in Nx takes care of that broadcast
implicitly, as discussed above:
Nx.subtract(tensor, Nx.tensor([1, 2]))
The broadcast worked as expected, copying the [1, 2] row
enough times to fill a {2, 2}-shaped tensor. A tensor with a
dimension of 1 will broadcast to fill the tensor:
[[1], [2]] |> Nx.tensor() |> Nx.broadcast({1, 2, 2})
[[[1, 2, 3]]]
|> Nx.tensor()
|> Nx.broadcast({4, 2, 3})
Both of these examples copy parts of the tensor enough
times to fill out the broadcast shape. You can check out the
Nx broadcasting documentation for more details:
h Nx.broadcast
Much of the time, you won't have to broadcast yourself. Many of
the functions and operators Nx supports will do so automatically.
We can use tensor-aware operators via various Nx functions and
many of them implicitly broadcast tensors.
Throughout this section, we have been invoking Nx.subtract/2 and
our code would be more expressive if we could use its equivalent
mathematical operator. Fortunately, Nx provides a way. Next, we'll
dive into numerical definitions using defn.


  

    
Numerical Definitions (defn)
    


  
    
  
  Section


The defn macro and its siblings simplify the expression of mathematical formulas
containing tensors. Numerical definitions have two primary benefits
over classic Elixir functions.
	They are tensor-aware. Nx replaces operators like Kernel.-/2
with the Defn counterpart, imported from Nx.Defn.Kernel which uses Nx functions
optimized for tensors when the operands are tensors, so the formulas we express can use
tensors out of the box.

	defn definitions allow for building computation graphs that combine the
individual operations, which can then be used by just-in-time (JIT) compilers
to emit highly specialized native code for the desired computation unit.


We don't have to do anything special to get access to
get tensor awareness beyond importing Nx.Defn and writing
our code within a defn block.
To use Nx in a Mix project or a notebook, we need to include
the :nx dependency and import the Nx.Defn module,
like this:
Mix.install([
  {:nx, "~> 0.9"}
])
import Nx.Defn
Just as the Elixir language supports def, defmacro, and defp,
Nx supports defn. There are a few restrictions. It allows only
numerical arguments in the form of primitives or tensors as arguments
or return values, and supports only a subset of the language.
The subset of Elixir allowed within defn is quite broad, though. We can
use macros, pipes, and even conditionals, so we're not giving up
much when you're declaring mathematical functions.
Additionally, despite these small concessions, defn provides huge benefits.
Code inside a defn block uses tensor-aware operators and types, so the math
beneath your functions has a better chance to shine through. Numerical
definitions can also run on accelerated numerical processors like GPUs and
TPUs. Here's an example numerical definition:
defmodule TensorMath do
  import Nx.Defn

  defn subtract(a, b) do
    a - b
  end
end
This module has a numerical definition that will be compiled.
If we wanted to specify a compiler for this module, we could add
a module attribute before the defn clause. One of such compilers
is the EXLA compiler.
You would add the mix dependency for EXLA and do this:
Nx.Defn.compile(&TensorMath.subtract/2, [Nx.template({3}, :f32), Nx.template({3}, :s32)], compiler: EXLA)
For a global approach, you can also set config :nx, default_defn_options: [compiler: EXLA]
in your application environment and call the TensorMath.subtract/2 function directly.
As an exercise, you can try adding a defn to TensorMath
that accepts two tensors representing the lengths of sides of a
right triangle and uses the pythagorean theorem to return the
length of the hypotenuse.
Add your function directly to the previous Code cell.

  
    
  
  deftransform


The defn macro in Nx allows you to define functions that compile to efficient
numerical computations, but it comes with certain limitations — such as restrictions
on argument types, return values, and the subset of Elixir that it supports.
To overcome many of these limitations, Nx offers the deftransform macro.
deftransform lets you perform computations or execute code that isn't directly
supported by defn, and then incorporate those results back into your numerical
function. This separation lets you use standard Elixir features where necessary
while keeping your core numerical logic optimized.
It is important to highlight that all code inside the deftransform call is being invoked
during the Nx compilation step, and thus can't depend on runtime values of the tensor inputs.
In the following example, we define a deftransform function called
compute_tensor_from_list/1 that receives a list, which is not allowed
inside defn. Inside this function, we convert the list to a tensor
using Nx.tensor/1, and then pass it to a defn function called double_tensor/1,
which performs the actual numerical computation.
defmodule MyMath do
  import Nx.Defn

  # Numerical function that just multiplies the tensor by a scalar
  defn scale_tensor(tensor) do
    Nx.multiply(tensor, 10)
  end

  # This transform receives a 2D list, validates it, reshapes it,
  # adds a new axis, and then passes it to a numerical function.
  deftransform compute_from_2d_list(list_2d) do
    # Validate that it's a proper matrix (all rows same length)
    lengths = Enum.map(list_2d, &length/1)
    if Enum.uniq(lengths) != [hd(lengths)] do
      raise ArgumentError, "All inner lists must have the same length"
    end

    # Convert to tensor (e.g., shape {2, 3})
    tensor = Nx.tensor(list_2d)

    # Add a new axis at the beginning: {2, 3} -> {1, 2, 3}
    reshaped = Nx.new_axis(tensor, 0)

    # Pass to defn function
    scale_tensor(reshaped)
  end
end
matrix = [
  [1, 2, 3],
  [4, 5, 6]
]

result = MyMath.compute_from_2d_list(matrix)
This setup allows us to keep our defn code clean and focused only on tensor
operations, while using deftransform to handle Elixir-native types and
preprocessing.
deftransforms can also be called from within defn functions, which
can be very useful for shape manipulation and validation:
defmodule MyOtherMath do
  import Nx.Defn

  # Numerical function that just multiplies the tensor by a scalar
  defn reshape_and_scale_tensor(tensor, opts \\ []) do
    tensor
    |> validate_and_reshape(opts[:new_axes_count])
    |> Nx.multiply(10)
  end

  deftransformp validate_and_reshape(tensor, new_axes_count) do
    cond do
      nil ->
        tensor

      is_integer(new_axes_count) ->
        shape = List.to_tuple(Nx.shape(tensor) ++ List.duplicate(1, new_axes_count))
        Nx.reshape(tensor, shape)

      true ->
        raise ArgumentError, "expected :new_axes_count to be an integer or nil, got: #{inspect(new_axes_count)}"
    end
  end
end


  

    
NumPy -&gt; Nx
    

This cheatsheet is designed to assist Python developers in transitioning to Elixir,
specifically by providing equivalent commands and code examples between NumPy and Nx.

  
    
  
  Tensor Creation



  
    
  
  From list or nested list


NumPy
>>> np.array([1, 2, 3])
array([1, 2, 3])
Nx
iex> Nx.tensor([1, 2, 3])
#Nx.Tensor<
  s32[3]
  [1, 2, 3]
>

  
    
  
  2D Arrays/Tensors


NumPy
>>> np.array([[1, 2], [3, 4]])
array([[1, 2],
       [3, 4]])
Nx
iex> Nx.tensor([[1, 2], [3, 4]])
#Nx.Tensor<
  s32[2][2]
  [
    [1, 2],
    [3, 4]
  ]
>

  
    
  
  Zeros and Ones


NumPy
>>> np.zeros((2, 3))
array([[0., 0., 0.],
       [0., 0., 0.]])

>>> np.ones((2, 3))
array([[1., 1., 1.],
       [1., 1., 1.]])
NumPy
iex> Nx.broadcast(0, {2, 3})
#Nx.Tensor<
  s32[2][3]
  [
    [0, 0, 0],
    [0, 0, 0]
  ]
>

iex> Nx.broadcast(1, {2, 3})
#Nx.Tensor<
  s32[2][3]
  [
    [1, 1, 1],
    [1, 1, 1]
  ]
>

  
    
  
  Range of Numbers


NumPy
>>> np.arange(0, 10, 2)
array([0, 2, 4, 6, 8])
NumPy
iex> Nx.iota({5}, axis: 0) |> Nx.multiply(2)
#Nx.Tensor<
  s32[5]
  [0, 2, 4, 6, 8]
>

  
    
  
  Linearly Spaced Values


NumPy
>>> np.linspace(0, 1, 5)
array([0.  , 0.25, 0.5 , 0.75, 1.  ])
NumPy
iex> Nx.iota({5}) |> Nx.divide(4)
#Nx.Tensor<
  f32[5]
  [0.0, 0.25, 0.5, 0.75, 1.0]
>

  
    
  
  Tensor Inspection



  
    
  
  Shape


NumPy
>>> a = np.array([[1, 2, 3], [4, 5, 6]])
>>> a.shape
(2, 3)
Nx
iex> a = Nx.tensor([[1, 2, 3], [4, 5, 6]])
#Nx.Tensor<
  s32[2][3]
  [
    [1, 2, 3],
    [4, 5, 6]
  ]
>
iex> Nx.shape(a)
{2, 3}

  
    
  
  Number of dimensions


NumPy
>>> a.ndim
2
Nx
iex> Nx.rank(a)
2

  
    
  
  Data Type


NumPy
>>> a.dtype
dtype('int64')
Nx
iex> Nx.type(a)
{:s, 32}

  
    
  
  Total Number of Elements


NumPy
>>> a.size
6
Nx
iex> Nx.size(a)
6

  
    
  
  Indexing and Slicing



  
    
  
  Indexing a Single Element


NumPy
>>> a = np.array([[10, 20], [30, 40]])
>>> a[0, 1]
np.int64(20)
Nx
# Indexing a Single Element
iex> tensor = Nx.tensor([[10, 20], [30, 40]])
iex> tensor[[0, 1]]
#Nx.Tensor<
  s32
  20
>

  
    
  
  Slicing a Range


NumPy
>>> a = np.array([10, 20, 30, 40, 50])
>>> a[1:4]
array([20, 30, 40])
Nx
# Slicing a Range
iex> a = Nx.tensor([10, 20, 30, 40, 50])
iex> a[1..3]
#Nx.Tensor<
  s32[3]
  [20, 30, 40]
>

  
    
  
  Selecting Along a Specific Axis


NumPy
>>> a = np.array([[1, 2], [3, 4], [5, 6]])
>>> a[:, 1]
array([2, 4, 6])
Nx
# Selecting Along a Specific Axis
iex> a = Nx.tensor([[1, 2], [3, 4], [5, 6]])
iex> a[[.., 1]]
#Nx.Tensor<
  s32[3]
  [2, 4, 6]
>

  
    
  
  Boolean Masking


NumPy
>>> x = np.arange(10)
>>> x[x % 2 == 0]
array([0, 2, 4, 6, 8])
Nx
Boolean masking requires dynamic shape behavior, which is not
supported in Nx because Nx compiles all operations
ahead-of-time (like XLA or Jax), and for that, tensors must have static shapes.

  
    
  
  Linear Algebra Operations



  
    
  
  Matrix Multiplication


NumPy
>>> A = np.array([[1, 2], [3, 4]])
>>> B = np.array([[5, 6], [7, 8]])
>>> np.matmul(A, B)
array([[19, 22],
       [43, 50]])
Nx
iex> a = Nx.tensor([[1, 2], [3, 4]])
iex> b = Nx.tensor([[5, 6], [7, 8]])
iex> Nx.dot(a, b)
#Nx.Tensor<
  s32[2][2]
  [
    [19, 22],
    [43, 50]
  ]
>

  
    
  
  Transpose


NumPy
>>> A.T
array([[1, 3],
       [2, 4]])
Nx
iex> Nx.transpose(a)
#Nx.Tensor<
  s32[2][2]
  [
    [1, 3],
    [2, 4]
  ]
>

  
    
  
  Identity Matrix


NumPy
>>> np.eye(3)
array([[1., 0., 0.],
       [0., 1., 0.],
       [0., 0., 1.]])
Nx
iex> Nx.eye({3, 3})
#Nx.Tensor<
  s32[3][3]
  [
    [1, 0, 0],
    [0, 1, 0],
    [0, 0, 1]
  ]
>

  
    
  
  Determinant


NumPy
>>> np.linalg.det(A)
np.float64(-2.0000000000000004)
Nx
iex> Nx.LinAlg.determinant(a)
#Nx.Tensor<
  f32
  -2.0
>

  
    
  
  Inverse


NumPy
>>> np.linalg.inv(A)
array([[-2. ,  1. ],
       [ 1.5, -0.5]])
Nx
iex> Nx.LinAlg.invert(a)
#Nx.Tensor<
  f32[2][2]
  [
    [-2.000000476837158, 1.0000003576278687],
    [1.5000004768371582, -0.5000002384185791]
  ]
>

  
    
  
  Solve a System of Linear Equations


NumPy
>>> A = np.array([[3, 1], [1, 2]])
>>> b = np.array([9, 8])
>>> np.linalg.solve(A, b)
array([2., 3.])
Nx
iex> a = Nx.tensor([[3.0, 1.0], [1.0, 2.0]])
iex> b = Nx.tensor([9.0, 8.0])
iex> Nx.LinAlg.solve(a, b)
#Nx.Tensor<
  f32[2]
  [2.0, 3.0]
>

  
    
  
  Eigenvalues and Eigenvectors


NumPy
>>> np.linalg.eigh(A)
EighResult(
          eigenvalues=array([1.38196601, 3.61803399]),
          eigenvectors=array([
                              [ 0.52573111, -0.85065081],
                              [-0.85065081, -0.52573111]
                              ]))
Nx
iex> Nx.LinAlg.eigh(a)
{#Nx.Tensor<
   f32[2]
   [3.618025779724121, 1.381974220275879]
 >,
 #Nx.Tensor<
   f32[2][2]
   [
     [0.8516583442687988, -0.5240974426269531],
     [0.5240974426269531, 0.8516583442687988]
   ]
 >}


  

    
Exercises: 1-20
    

Mix.install([{:nx, "~> 0.6"}])

  
    
  
  Introduction


Inspired by the Python notebook 100 Numpy Exercises.

  
    
  
  1-10


1. Install Nx in a Livebook. (★☆☆)
# Add your solution here.
  Example solution
  

  ```elixir
  Mix.install([{:nx, "~> 0.6"}])
  ```

  
2. Create a 1-D tensor of 10 zeros. (★☆☆)
# Add your solution here.
  Example solution
  

  ```elixir
  Nx.broadcast(0, {10})
  ```

  
3. Find the number of elements in tensor. (★☆☆)
tensor = Nx.tensor([[1, 2, 3], [4, 5, 6]])
# Add your solution here.
  Example solution
  

  ```elixir
  Nx.size(tensor)
  ```

  
4. Find the number of bytes of memory in tensor. (★☆☆)
tensor = Nx.tensor([[1, 2, 3], [4, 5, 6]])
# Add your solution here.
  Example solution
  

  ```elixir
  Nx.byte_size(tensor)
  ```

  
5a. Use Nx.sum/2 to find the sum of all elements of tensor. (★☆☆)
tensor = Nx.tensor([[1, 2, 3], [4, 5, 6], [7, 8, 9]])
# Add your solution here.
  Example solution
  

  ```elixir
  Nx.sum(tensor)
  ```

  
5b. Read the documentation for Nx.sum/2 then provide the correct option to sum across the rows. (★☆☆)
# Add your solution here.
  Example solution
  

  ```elixir
  Nx.sum(tensor, axes: [1])
  ```

  
    Tip: You can also hover over a function inside Livebook code cells to display its documentation.
  


  
6. Create a tensor of zeros of size 10 but where the fifth value is 1. (★☆☆)
# Add your solution here.
  Example solution
  

  ```elixir
  zeros = Nx.broadcast(0, {10})
  index = Nx.tensor([4])
  Nx.indexed_put(zeros, index, 1)
  ```

  
7. Create a 3x3 tensor with values ranging from 0 to 8. (★☆☆)
# Add your solution here.
  Example solution
  

  ```elixir
  Nx.iota({3, 3})
  ```

  
8. Create a tensor with values ranging from 10 to 49. (★☆☆)
# Add your solution here.
  Example solution 1
  

  ```elixir
  Nx.iota({40})
  |> Nx.add(10)
  ```

  
  Example solution 2
  

  ```elixir
  Nx.linspace(10, 49, n: 39, type: :s64)
  ```

  
9. Reverse tensor (first element becomes last). (★☆☆)
tensor = Nx.tensor([2, 4, 6, 8])
# Add your solution here.
  Example solution
  

  ```elixir
  Nx.reverse(tensor)
  ```

  
10a. Given an initial tensor, build a "mask" of non-zero elements. That is, build a second tensor with the same shape as the original, but that has a 1 wherever the original has a non-zero element and a 0 elsewhere. (★☆☆)
tensor = Nx.tensor([1, 2, 0, 0, 4, 0])
# Add your solution here.
  Example solution
  

  ```elixir
  mask = Nx.not_equal(tensor, 0)
  ```

  
10b. Use the mask from 10a to replace each 0 from tensor with -1. (★☆☆)
# Add your solution here.
  Example solution
  

  ```elixir
  Nx.select(mask, tensor, -1)
  ```

  

  
    
  
  11-20


11. Create a 3x3 identity tensor. (★☆☆)
# Add your solution here.
  Example solution
  

  ```elixir
  Nx.eye(3)
  ```

  
12. Create a 3x3x3 tensor with random values. (★☆☆)
# Add your solution here.
  Example solution
  

  ```elixir
  key = Nx.Random.key(0)
  {random, _} = Nx.Random.normal(key, shape: {3, 3, 3})
  random
  ```

  
13. Create a random 10x10 tensor then find its minimum and maximum values. (★☆☆)
# Add your solution here.
  Example solution
  

  ```elixir
  key = Nx.Random.key(0)
  {tensor, _} = Nx.Random.normal(key, shape: {10, 10})

  %{
    min: Nx.reduce_min(tensor),
    max: Nx.reduce_max(tensor)
  }
  ```

  
14. Create a random 1D tensor of size 30 then find its mean. (★☆☆)
# Add your solution here.
  Example solution
  

  ```elixir
  key = Nx.Random.key(0)
  {tensor, _} = Nx.Random.normal(key, shape: {30})

  Nx.mean(tensor)
  ```

  
15. Create a 4x4 tensor with 1 on the border and 0 inside. (★☆☆)
# Add your solution here.
  Example solution
  

  ```elixir
  Nx.broadcast(1, {4, 4})
  |> Nx.put_slice([1, 1], Nx.broadcast(0, {2, 2}))
  ```

  
16. Add a border of 0 around tensor (end result will be a 5x5 tensor). (★☆☆)
tensor = Nx.broadcast(1, {3, 3})
# Add your solution here.
  Example solution
  

  ```elixir
  Nx.pad(tensor, 0, [{1, 1, 0}, {1, 1, 0}])
  ```

  
17. Determine the results of the following expressions. (★☆☆)
nan = Nx.Constants.nan()
Nx.multiply(0, nan)
Nx.equal(nan, nan)
Nx.greater(nan, nan)
Nx.subtract(nan, nan)

# Add your solution here.
  Example solution
  

  ```
  #Nx.Tensor

  {:nx, "~> 0.7"}

  What is vectorization?


Vectorization in Nx is the concept of imposing "for-each" semantics into leading tensor axes. This enables writing tensor operations in a simpler way, avoiding while loops, which also leading to more efficient code, as we'll see in this guide.
There are a few functions related to vectorization. First we'll look into Nx.vectorize/2, Nx.devectorize/2, and Nx.revectorize/3.
For these examples, we'll utilize the normalize function defined below, which only works with 1D tensors. The function subtracts the minimum value and divides the tensor by the resulting maximum.
defmodule Example do
  import Nx.Defn

  defn normalize(t) do
    case Nx.shape(t) do
      {_} -> :ok
      _ -> raise "invalid shape"
    end

    min = Nx.reduce_min(t)
    zero_min = Nx.subtract(t, min)
    Nx.divide(zero_min, Nx.reduce_max(zero_min))
  end
end
{:module, Example, <<70, 79, 82, 49, 0, 0, 11, ...>>, true}
We can invoke it as:
Example.normalize(Nx.tensor([1, 2, 3]))
#Nx.Tensor<
  f32[3]
  [0.0, 0.5, 1.0]
>
However, if we attempt to call it for any tensor with more than one dimension, it won't work. Let's first define a 2D tensor:
t =
  Nx.tensor([
    [1, 2, 3],
    [10, 20, 30],
    [4, 5, 6]
  ])
#Nx.Tensor<
  s64[3][3]
  [
    [1, 2, 3],
    [10, 20, 30],
    [4, 5, 6]
  ]
>
Now if we attempt to invoke it:
Example.normalize(t)
To address this, we can vectorize the tensor:
vec = Nx.vectorize(t, :rows)
#Nx.Tensor<
  vectorized[rows: 3]
  s64[3]
  [
    [1, 2, 3],
    [10, 20, 30],
    [4, 5, 6]
  ]
>
As we can see above, the newly vectorized vec is the same tensor as t, but the first axis is now a vectorized axis called rows, with size 3. This means that for all intents and purposes, we can think of this tensor as a 1D tensor, on which our normalize function will now work as if we passed those 3 rows separately (thus the "for-each" semantics mentioned above). Let's give it a try:
normalized_vec = Example.normalize(vec)
#Nx.Tensor<
  vectorized[rows: 3]
  f32[3]
  [
    [0.0, 0.5, 1.0],
    [0.0, 0.5, 1.0],
    [0.0, 0.5, 1.0]
  ]
>
While the tensor is vectorized, we can't treat it as a matrix (2D tensor):
right =
  Nx.tensor([
    [1, 2, 3],
    [2, 3, 4],
    [3, 4, 5]
  ])

# The results might be unexpected, but they behave the same as Nx.add(Nx.tensor([1, 2, 3]), right)
# and so on, resulting in a vectorized tensor with 3 inner matrices.
Nx.add(normalized_vec, right)
#Nx.Tensor<
  vectorized[rows: 3]
  f32[3][3]
  [
    [
      [1.0, 2.5, 4.0],
      [2.0, 3.5, 5.0],
      [3.0, 4.5, 6.0]
    ],
    [
      [1.0, 2.5, 4.0],
      [2.0, 3.5, 5.0],
      [3.0, 4.5, 6.0]
    ],
    [
      [1.0, 2.5, 4.0],
      [2.0, 3.5, 5.0],
      [3.0, 4.5, 6.0]
    ]
  ]
>
You can devectorize the tensor to get its original shape:
# If we want to keep the vectorized axes' names
Nx.devectorize(normalized_vec)
#Nx.Tensor<
  f32[rows: 3][3]
  [
    [0.0, 0.5, 1.0],
    [0.0, 0.5, 1.0],
    [0.0, 0.5, 1.0]
  ]
>
# If we want to drop the vectorized axes' names
devec = Nx.devectorize(normalized_vec, keep_names: false)
#Nx.Tensor<
  f32[3][3]
  [
    [0.0, 0.5, 1.0],
    [0.0, 0.5, 1.0],
    [0.0, 0.5, 1.0]
  ]
>
Once devectorized, we can effectively add the two matrices together:
Nx.add(devec, right)
#Nx.Tensor<
  f32[3][3]
  [
    [1.0, 2.5, 4.0],
    [2.0, 3.5, 5.0],
    [3.0, 4.5, 6.0]
  ]
>

  
    
  
  Revectorization


Now that we have the basics down, let's discuss Nx.revectorize with multi dimensional tensors. This is especially useful in cases where we have multiple vectorization axes which we want to collapse and then re-expand.
# This version of vectorize also asserts on the size of the dimensions being vectorized
t = Nx.iota({2, 1, 3, 2, 2}) |> Nx.vectorize(x: 2, y: 1, z: 3)
#Nx.Tensor<
  vectorized[x: 2][y: 1][z: 3]
  s64[2][2]
  [
    [
      [
        [
          [0, 1],
          [2, 3]
        ],
        [
          [4, 5],
          [6, 7]
        ],
        [
          [8, 9],
          [10, 11]
        ]
      ]
    ],
    [
      [
        [
          [12, 13],
          [14, 15]
        ],
        [
          [16, 17],
          [18, 19]
        ],
        [
          [20, 21],
          [22, 23]
        ]
      ]
    ]
  ]
>
Let's imagine we want to operate on the vectorized iota t above, adding a specific constant for each vectorized entry. We can do this through a bit of tensor introspection in conjunction with revectorize/3
collapsed_t = Nx.revectorize(t, [vectors: :auto], target_shape: t.shape)

[vectors: vectorized_size] = collapsed_t.vectorized_axes

constants =
  Nx.iota({vectorized_size})
  |> Nx.multiply(100)
  |> Nx.vectorize(vectors: vectorized_size)

{constants, collapsed_t}
{#Nx.Tensor<
   vectorized[vectors: 6]
   s64
   [0, 100, 200, 300, 400, 500]
 >,
 #Nx.Tensor<
   vectorized[vectors: 6]
   s64[2][2]
   [
     [
       [0, 1],
       [2, 3]
     ],
     [
       [4, 5],
       [6, 7]
     ],
     [
       [8, 9],
       [10, 11]
     ],
     [
       [12, 13],
       [14, 15]
     ],
     [
       [16, 17],
       [18, 19]
     ],
     [
       [20, 21],
       [22, 23]
     ]
   ]
 >}
From the output above, we can see that we have a vectorized tensor of scalars as well a vectorized tensor with the same vectorized axis, containing matrices.
Now we can add them together as we discussed above, and then re-expand the vectorized axes to the original shape.
collapsed_t
|> Nx.add(constants)
|> Nx.revectorize(t.vectorized_axes)
#Nx.Tensor<
  vectorized[x: 2][y: 1][z: 3]
  s64[2][2]
  [
    [
      [
        [
          [0, 1],
          [2, 3]
        ],
        [
          [104, 105],
          [106, 107]
        ],
        [
          [208, 209],
          [210, 211]
        ]
      ]
    ],
    [
      [
        [
          [312, 313],
          [314, 315]
        ],
        [
          [416, 417],
          [418, 419]
        ],
        [
          [520, 521],
          [522, 523]
        ]
      ]
    ]
  ]
>
Before we move on to the last two vectorization functions, let's see how we can replace a while loop with vectorization. The following module defines the same function twice, once for each method.
defmodule WhileExample do
  import Nx.Defn

  defn while_sum_and_multiply(x, y) do
    out =
      case Nx.shape(x) do
        {n, _} ->
          Nx.broadcast(0, {n})

        _ ->
          raise "expected x to have rank 2"
      end

    n = Nx.axis_size(out, 0)

    case Nx.shape(y) do
      {^n} -> nil
      _ -> raise "expected y to have rank 1 and the same number of elements as x"
    end

    {out, _} =
      while {out, {x, y}}, i <- 0..(n - 1) do
        update = Nx.sum(x[i]) + y[i]

        updated = Nx.indexed_put(out, Nx.reshape(i, {1}), update)
        {updated, {x, y}}
      end

    out
  end

  defn vectorized_sum_and_multiply(x, y) do
    # for the sake of equivalence, we'll keep the limitation
    # of accepting only rank-2 tensors for x
    {n, m} =
      case Nx.shape(x) do
        {n, m} ->
          {n, m}

        _ ->
          raise "expected x to have rank 2"
      end

    case Nx.shape(y) do
      {^n} -> nil
      _ -> raise "expected y to have rank 1 and the same number of elements as x"
    end

    # this enables us to accept vectorized tensors for x and y
    vectorized_axes = x.vectorized_axes

    x = Nx.revectorize(x, [collapsed: :auto, vectors: n], target_shape: {m})
    y = Nx.revectorize(y, [collapsed: :auto, vectors: n], target_shape: {})

    out = Nx.sum(x) + y

    Nx.revectorize(out, vectorized_axes, target_shape: {n})
  end
end
{:module, WhileExample, <<70, 79, 82, 49, 0, 0, 22, ...>>, true}
Let's give those definitions a try:
x =
  Nx.tensor([
    [1, 2, 3],
    [1, 0, 0],
    [0, 1, 3]
  ])

y = Nx.tensor([4, 5, 6])

x_vec = Nx.tile(x, [2, 1]) |> Nx.reshape({2, 3, 3}) |> Nx.vectorize(:rows)
# use a different set of vectorized axes for y
y_vec = Nx.concatenate([y, Nx.multiply(y, 2)]) |> Nx.reshape({2, 3}) |> Nx.vectorize(:cols)

{
  WhileExample.while_sum_and_multiply(x, y),
  WhileExample.vectorized_sum_and_multiply(x, y),
  WhileExample.vectorized_sum_and_multiply(x_vec, y),
  WhileExample.vectorized_sum_and_multiply(x_vec, y_vec)
}
{#Nx.Tensor<
   s64[3]
   [10, 6, 10]
 >,
 #Nx.Tensor<
   s64[3]
   [10, 6, 10]
 >,
 #Nx.Tensor<
   vectorized[rows: 2]
   s64[3]
   [
     [10, 6, 10],
     [10, 6, 10]
   ]
 >,
 #Nx.Tensor<
   vectorized[rows: 2]
   s64[3]
   [
     [10, 6, 10],
     [14, 11, 16]
   ]
 >}
The advantage of vectorization is that the underlying compilers and hardware may do a much better job of optimizing tensor operations than while loops, which often run linearly.
The eagle-eyed reader will have noticed that we inadvertently threw away the vectorized axes that we received from y. This is because our usage of revectorize disregards the possibility that the axes could be different in each input tensor. Luckily, we can tackle this problem too.

  
    
  
  Reshape and Broadcast Vectors


The bug introduced in the previous section can be solved through Nx.reshape_vectors/1 and Nx.broadcast_vectors/1. Both functions receive lists of tensors and will operate on their vectorized axes to ensure that the shapes are compatible in one way or the other.
Nx functions will natively do this for us, as we see below:
base = Nx.tensor([0, 1, 2, 3, 4, 5, 6, 7, 8, 9])
vec_i = Nx.vectorize(Nx.reshape(base, {10, 1}), i: 10, j: 1)
vec_j = Nx.vectorize(base, :j)

# easy way to build a multiplication table from 0 to 9
Nx.multiply(vec_i, vec_j)
#Nx.Tensor<
  vectorized[i: 10][j: 10]
  s64
  [
    [0, 0, 0, 0, 0, 0, 0, 0, 0, 0],
    [0, 1, 2, 3, 4, 5, 6, 7, 8, 9],
    [0, 2, 4, 6, 8, 10, 12, 14, 16, 18],
    [0, 3, 6, 9, 12, 15, 18, 21, 24, 27],
    [0, 4, 8, 12, 16, 20, 24, 28, 32, 36],
    ...
  ]
>
Ok, now that we know that broadcasting will work on vectorized axes to add elements in our resulting tensor, we can look into the aforementioned functions.
a = Nx.iota({2, 1}, vectorized_axes: [x: 2, y: 1])
b = Nx.iota({}, vectorized_axes: [z: 2, x: 1])

Nx.reshape_vectors([a, b], align_ranks: true)
[
  #Nx.Tensor<
    vectorized[x: 2][y: 1][z: 1]
    s64[2][1]
    [
      [
        [
          [
            [0],
            [1]
          ]
        ]
      ],
      [
        [
          [
            [0],
            [1]
          ]
        ]
      ]
    ]
  >,
  #Nx.Tensor<
    vectorized[x: 1][y: 1][z: 2]
    s64[1][1]
    [
      [
        [
          [
            [0]
          ],
          [
            [0]
          ]
        ]
      ]
    ]
  >
]
In the example above, we can see that both tensors end up containing vectorized axes :x, :y, and :z. Furthermore, we can also see that the b tensor was rearranged so that the :x axis comes first. All tensors end up with the same ordering of axes, but the axes aren't resized in any way.
Finally, the align_ranks: true option is passed so that the inner shape (the non-vectorized part!) of both tensors ends up in a broadcastable shape with the same rank across all tensors. The example uses only 2 tensors, but the list can have arbitrary size.
Nx.broadcast_vectors/2 works similarly, except it also broadcasts the dimensions instead of simply reshaping:
x = Nx.iota({2, 1}, vectorized_axes: [x: 2, y: 1])
y = Nx.iota({}, vectorized_axes: [z: 2, x: 1])

Nx.broadcast_vectors([x, y])
[
  #Nx.Tensor<
    vectorized[x: 2][y: 1][z: 2]
    s64[2][1]
    [
      [
        [
          [
            [0],
            [1]
          ],
          [
            [0],
            [1]
          ]
        ]
      ],
      [
        [
          [
            [0],
            [1]
          ],
          [
            [0],
            [1]
          ]
        ]
      ]
    ]
  >,
  #Nx.Tensor<
    vectorized[x: 2][y: 1][z: 2]
    s64
    [
      [
        [0, 0]
      ],
      [
        [0, 0]
      ]
    ]
  >
]
The key difference is that all vectorized axes will end up with the same size in all resulting tensors, effectively behaving the same as Nx.broadcast does for non-vectorized shapes. Specifically, we can see that the x tensor gets the new z: 2 axis and y gets both x: 2, y: 1, where the :x axis was already present, but has now been resized.
With this knowledge we can rewrite our function without the bug, as follows:
defmodule BroadcastVectorsExample do
  import Nx.Defn

  defn vectorized_sum_and_multiply(x, y) do
    # for the sake of equivalence, we'll keep the limitation
    # of accepting only rank-2 tensors for x

    [x, y] = Nx.broadcast_vectors([x, y])

    {n, m} =
      case Nx.shape(x) do
        {n, m} ->
          {n, m}

        _ ->
          raise "expected x to have rank 2"
      end

    case Nx.shape(y) do
      {^n} -> nil
      _ -> raise "expected y to have rank 1 and the same number of elements as x"
    end

    # this enables us to accept vectorized tensors for x and y
    vectorized_axes = x.vectorized_axes

    x = Nx.revectorize(x, [collapsed: :auto, vectors: n], target_shape: {m})
    y = Nx.revectorize(y, [collapsed: :auto, vectors: n], target_shape: {})

    out = Nx.sum(x) + y

    Nx.revectorize(out, vectorized_axes, target_shape: {n})
  end
end
{:module, BroadcastVectorsExample, <<70, 79, 82, 49, 0, 0, 15, ...>>, true}
Let's give it once again another try:
x =
  Nx.tensor([
    [1, 2, 3],
    [1, 0, 0],
    [0, 1, 3]
  ])

y = Nx.tensor([4, 5, 6])

x_vec = Nx.tile(x, [2, 1]) |> Nx.reshape({2, 3, 3}) |> Nx.vectorize(:rows)
# use a different set of vectorized axes for y
y_vec = Nx.concatenate([y, Nx.multiply(y, 2)]) |> Nx.reshape({2, 3}) |> Nx.vectorize(:cols)

{
  BroadcastVectorsExample.vectorized_sum_and_multiply(x_vec, y),
  BroadcastVectorsExample.vectorized_sum_and_multiply(x_vec, y_vec)
}
{#Nx.Tensor<
   vectorized[rows: 2]
   s64[3]
   [
     [10, 6, 10],
     [10, 6, 10]
   ]
 >,
 #Nx.Tensor<
   vectorized[rows: 2][cols: 2]
   s64[3]
   [
     [
       [10, 6, 10],
       [14, 11, 16]
     ],
     [
       [10, 6, 10],
       [14, 11, 16]
     ]
   ]
 >}

Mix.install([
  {:nx, "~> 0.9"}
])
import Nx, only: :sigils
Nx

  
    
  
  What is aggregation?


Aggregation is the process of reducing a tensor to a single value or a smaller tensor by applying specific operations across its dimensions. You can apply aggregation functions on any tensor. The functions can be applied to the tensor as a whole or to a subsection of the tensor taken in an axis-wise fashion.
As a first example, let's take a 2D tensor of shape {2, 3}. Notice that we can name the axes. The elements of the tensor t are the t[i][j] when $i$ is the row, and $j$ the column.
m = Nx.tensor([[1, 2, 3], [4, 5, 6]], names: [:x, :y])
#Nx.Tensor<
  s64[x: 2][y: 3]
  [
    [1, 2, 3],
    [4, 5, 6]
  ]
>
You can get the maximum number in a tensor with Nx.reduce_max(m), returning a 0D tensor. With a reduction, we lose a dimension per axis reduced, and since we applied the reduction globally, we lose all the dimensions. It should return Nx.Tensor(6) here.
We can get the maximum number for each row with Nx.reduce_max(matrix, axes: [:y]), returning a 1D tensor of size 2. Why :y? Because for each row, we reduce along the :y axis.
We can get the maximum number for each column with Nx.reduce_max(matrix, axes: [:x]), returning a 1D tensor of size 3. For each column, we reduce along the axis :x.
max = Nx.reduce_max(m)
max_x = Nx.reduce_max(m, axes: [:x])
max_y = Nx.reduce_max(m, axes: [:y])
%{max: max, max_x: max_x, max_y: max_y}
%{
  max: #Nx.Tensor<
    s64
    6
  >,
  max_x: #Nx.Tensor<
    s64[y: 3]
    [4, 5, 6]
  >,
  max_y: #Nx.Tensor<
    s64[x: 2]
    [3, 6]
  >
}
Let's consider another example with Nx.weighted_mean. It supports full-tensor and per axis operations. We display how to compute the weighted mean aggregate of a matrix with the example below of a 2D tensor of shape {2,2} labeled m:
m = ~MAT[
  1 2
  3 4
]
#Nx.Tensor<
  s64[2][2]
  [
    [1, 2],
    [3, 4]
  ]
>
First, we'll compute the full-tensor aggregation. The calculations are developed below. We calculate an "array product" (aka 

Mix.install([
  {:nx, "~> 0.7"}
])

  
    
  
  What is Function Differentiation?


Nx, through the Nx.Defn.grad/2 and Nx.Defn.value_and_grad/3 functions allows the user to differentiate functions that were defined through defn.
This is really important in Machine Learning settings because, in general, the training process happens through optimization methods that require calculating the gradient of tensor functions.
Before we get too far ahead of ourselves, let's talk about what is the derivative or the gradient of a function.
In simple terms, the derivative tells us how a function changes at a given point and lets us measure things such as where a function as maximum,
minimum or turning points (for example, where a parabola has its vertex).
The ability to measure local minima and maxima is what makes them important to optimization problems, because if we can find them, we can solve problems that want
to minimize a given function. For higher dimensional problems, we deal with functions of many variables, and thus we use the gradient, which measures the "derivative" in the axis of each function.
The gradient, then, is a vector that points in the direction where the function changes the most, which leads to the so-called gradient descent method of optimization.
In the gradient descent method, we take tiny steps following the gradient of the function in order to find the nearest local minimum (which hopefully is either the global minimum or close enough to it).
This is what makes function differentiation so important for Machine Learning.
Let's take for example the following $f(x)$ and $f'(x)$ scalar function and derivative pair:
$$
f(x) = x^3 + x\\
f'(x) = 3x^2 + 1
$$
We can define a similar function-derivative pair for tensor functions:
$$
f(\bold{x}) = \bold{x}^3 + \bold{x}\\
\nabla f(\bold{x}) = 3 \bold{x} ^ 2 + 1
$$
These may look similar, but the difference is that $f(\bold{x})$ takes in $\bold{x}$ which is a tensor argument. This means that we can have the following argument and results for the function and its gradient:
$$
\bold{x} =
\begin{bmatrix}
1 & 1 \\
2 & 3 \\
5 & 8 \\
\end{bmatrix}\\\
$$
$$
f(\bold{x}) = \bold{x}^3 + \bold{x} =
\begin{bmatrix}
2 & 2 \\
10 & 30 \\
130 & 520
\end{bmatrix}
$$
$$
\nabla f(\bold{x}) = 3 \bold{x} ^ 2 + 1 =
\begin{bmatrix}
4 & 4 \\
13 & 28 \\
76 & 193
\end{bmatrix}
$$

  
    
  
  Automatic Differentiation


Now that we have a general feeling of what a function and its gradient are, we can talk about how Nx can use defn to calculate gradients for us.
In the following code blocks we're going to define the same tensor function as above and then we'll differentiate it only using Nx, without having to write the explicit derivative at all.
defmodule Math do
  import Nx.Defn

  defn f(x) do
    x ** 3 + x
  end

  defn grad_f(x) do
    Nx.Defn.grad(x, &f/1)
  end
end
x =
  Nx.tensor([
    [1, 1],
    [2, 3],
    [5, 8]
  ])

{
  Math.f(x),
  Math.grad_f(x)
}
As we can see, we get the results we expected, aside from the type of the grad, which will always be a floating-point number, even if you pass an integer tensor as input.
Next, we'll using Nx.Defn.debug_expr to see what's happening under the hood.
Nx.Defn.debug_expr(&Math.f/1).(x)
Nx.Defn.debug_expr(&Math.grad_f/1).(x)
If we look closely at the returned Nx.Defn.Expr representations for f and grad_f, we can see that they pretty much translate to the mathematical definitions we had originally.
This possible because Nx holds onto the symbolic representation of a defn function while inside defn-land, and thus Nx.Defn.grad (and similar) can operate on that symbolic representation to return a new symbolic representation (as seen in the second block).
Nx.Defn.value_and_grad can be used to calculate both things at once for us:
Nx.Defn.value_and_grad(x, &Math.f/1)
And if we use debug_expr again, we can see that the symbolic representation is actually both the function and the grad, returned in a tuple:
Nx.Defn.debug_expr(Nx.Defn.value_and_grad(&Math.f/1)).(x)
Finally, we can talk about functions that receive many arguments, such as the following add_multiply function:
add_multiply = fn x, y, z ->
  addition = Nx.add(x, y)
  Nx.multiply(z, addition)
end
At first you may think that if we want to differentiate it, we need to wrap it into a single-argument function so that we can differentiate with respect to a specific argument, which would treat other arguments as constants, as we can see below:
x = Nx.tensor([1, 2])
y = Nx.tensor([3, 4])
z = Nx.tensor([5, 6])

{
  Nx.Defn.grad(x, fn t -> add_multiply.(t, y, z) end),
  Nx.Defn.grad(y, fn t -> add_multiply.(x, t, z) end),
  Nx.Defn.grad(z, fn t -> add_multiply.(x, y, t) end)
}
However, Nx is smart enough to deal with multi-valued functions through Nx.Container representations such as a tuple or a map:
Nx.Defn.grad({x, y, z}, fn {x, y, z} -> add_multiply.(x, y, z) end)
Likewise, we can also deal with functions that return multiple values.
Nx.Defn.grad requires us to return a scalar from function (that is, a tensor of shape {}).
However, there are instances where we might want to use value_and_grad to get out a tuple from our function, while still calculating its gradient.
For this, we have the value_and_grad/3 arity, which accepts a transformation argument.
x =
  Nx.tensor([
    [1, 1],
    [2, 3],
    [5, 8]
  ])

# Notice that the returned values are the 2 addition terms from `Math.f/1`
multi_valued_return_fn =
  fn x ->
    {Nx.pow(x, 3), x}
  end

transform_fn = fn {x_cubed, x} -> Nx.add(x_cubed, x) end

{{x_cubed, x}, grad} = Nx.Defn.value_and_grad(x, multi_valued_return_fn, transform_fn)
If we go back to the start of this livebook, we can see that grad holds exactly the result Math.grad_f, but now we have access to x ** 3, which wasn't accessible before, as originally we could only obtain x ** 3 + x.

Numerical Elixir.
The Nx library is a collection of functions and data
types to work with Numerical Elixir. This module defines
the main entry point for building and working with said
data-structures. For example, to create an n-dimensional
tensor, do:
iex> t = Nx.tensor([[1, 2], [3, 4]])
iex> Nx.shape(t)
{2, 2}
Nx also provides the so-called numerical definitions under
the Nx.Defn module. They are a subset of Elixir tailored for
numerical computations. For example, it overrides Elixir's
default operators so they are tensor-aware:
defn softmax(t) do
  Nx.exp(t) / Nx.sum(Nx.exp(t))
end
Code inside defn functions can also be given to custom compilers,
which can compile said functions just-in-time (JIT) to run on the
CPU or on the GPU.

  
    
  
  References


Here is a general outline of the main references in this library:
	For an introduction, see our Intro to Nx guide

	This module provides the main API for working with tensors

	Nx.Defn provides numerical definitions, CPU/GPU compilation, gradients, and more

	Nx.LinAlg provides functions related to linear algebra

	Nx.Constants declares many constants commonly used in numerical code


Continue reading this documentation for an overview of creating,
broadcasting, and accessing/slicing Nx tensors.

  
    
  
  Creating tensors


The main APIs for creating tensors are tensor/2, from_binary/2,
iota/2, eye/2, and broadcast/3.
The tensor types can be one of:
	unsigned integers (u2, u4, u8, u16, u32, u64)
	signed integers (s2, s4, s8, s16, s32, s64)
	floats (f8, f16, f32, f64)
	brain floats (bf16)
	and complex numbers (c64, c128)

The types are tracked as tuples:
iex> Nx.tensor([1, 2, 3], type: {:f, 32})
#Nx.Tensor<
  f32[3]
  [1.0, 2.0, 3.0]
>
But a shortcut atom notation is also available:
iex> Nx.tensor([1, 2, 3], type: :f32)
#Nx.Tensor<
  f32[3]
  [1.0, 2.0, 3.0]
>
The tensor dimensions can also be named, via the :names option
available to all creation functions:
iex> Nx.iota({2, 3}, names: [:x, :y])
#Nx.Tensor<
  s32[x: 2][y: 3]
  [
    [0, 1, 2],
    [3, 4, 5]
  ]
>
Finally, for creating vectors and matrices, a sigil notation
is available:
iex> import Nx, only: :sigils
iex> ~VEC[1 2 3]f32
#Nx.Tensor<
  f32[3]
  [1.0, 2.0, 3.0]
>

iex> import Nx, only: :sigils
iex> ~MAT'''
...> 1 2 3
...> 4 5 6
...> '''s32
#Nx.Tensor<
  s32[2][3]
  [
    [1, 2, 3],
    [4, 5, 6]
  ]
>
All other APIs accept exclusively numbers or tensors, unless
explicitly noted otherwise.

  
    
  
  Broadcasting


Broadcasting allows operations on two tensors of different shapes
to match. For example, most often operations between tensors have
the same shape:
iex> a = Nx.tensor([1, 2, 3])
iex> b = Nx.tensor([10, 20, 30])
iex> Nx.add(a, b)
#Nx.Tensor<
  s32[3]
  [11, 22, 33]
>
Now let's imagine you want to multiply a large tensor of dimensions
1000x1000x1000 by 2. If you had to create a similarly large tensor
only to perform this operation, it would be inefficient. Therefore,
you can simply multiply this large tensor by the scalar 2, and Nx
will propagate its dimensions at the time the operation happens,
without allocating a large intermediate tensor:
iex> Nx.multiply(Nx.tensor([1, 2, 3]), 2)
#Nx.Tensor<
  s32[3]
  [2, 4, 6]
>
In practice, broadcasting is not restricted only to scalars; it
is a general algorithm that applies to all dimensions of a tensor.
When broadcasting, Nx compares the shapes of the two tensors,
starting with the trailing ones, such that:
	If the dimensions have equal size, then they are compatible

	If one of the dimensions have size of 1, it is "broadcast"
to match the dimension of the other


In case one tensor has more dimensions than the other, the missing
dimensions are considered to be of size one. Here are some examples
of how broadcast would work when multiplying two tensors with the
following shapes:
s32[3] * s64
#=> s32[3]

s32[255][255][3] * s32[3]
#=> s32[255][255][3]

s32[2][1] * s[1][2]
#=> s32[2][2]

s32[5][1][4][1] * s32[3][4][5]
#=> s32[5][3][4][5]
If any of the dimensions do not match or are not 1, an error is
raised.

  
    
  
  Access syntax (slicing)


Nx tensors implement Elixir's access syntax. This allows developers
to slice tensors up and easily access sub-dimensions and values.
Access accepts integers:
iex> t = Nx.tensor([[1, 2], [3, 4]])
iex> t[0]
#Nx.Tensor<
  s32[2]
  [1, 2]
>
iex> t[1]
#Nx.Tensor<
  s32[2]
  [3, 4]
>
iex> t[1][1]
#Nx.Tensor<
  s32
  4
>
If a negative index is given, it accesses the element from the back:
iex> t = Nx.tensor([[1, 2], [3, 4]])
iex> t[-1][-1]
#Nx.Tensor<
  s32
  4
>
Out of bound access will raise:
iex> Nx.tensor([1, 2])[2]
** (ArgumentError) index 2 is out of bounds for axis 0 in shape {2}

iex> Nx.tensor([1, 2])[-3]
** (ArgumentError) index -3 is out of bounds for axis 0 in shape {2}
The index can also be another tensor. If the tensor is a scalar, it must
be a value between 0 and the dimension size, and it behaves the same as
an integer. Out of bound dynamic indexes are always clamped to the tensor
dimensions:
iex> two = Nx.tensor(2)
iex> t = Nx.tensor([[1, 2], [3, 4]])
iex> t[two][two]
#Nx.Tensor<
  s32
  4
>
For example, a minus_one dynamic index will be clamped to zero:
iex> minus_one = Nx.tensor(-1)
iex> t = Nx.tensor([[1, 2], [3, 4]])
iex> t[minus_one][minus_one]
#Nx.Tensor<
  s32
  1
>
A multi-dimensional tensor uses its values to fetch the leading
dimension of the tensor, placing them within the shape of the
indexing tensor. It is equivalent to take/3:
iex> t = Nx.tensor([[1, 2], [3, 4]])
iex> t[Nx.tensor([1, 0])]
#Nx.Tensor<
  s32[2][2]
  [
    [3, 4],
    [1, 2]
  ]
>
The example shows how the retrieved indexes are nested
with the accessed shape and that you may also access
repeated indices:
iex> t = Nx.tensor([[1, 2], [3, 4]])
iex> t[Nx.tensor([[1, 0, 1]])]
#Nx.Tensor<
  s32[1][3][2]
  [
    [
      [3, 4],
      [1, 2],
      [3, 4]
    ]
  ]
>
Access also accepts ranges. Ranges in Elixir are inclusive:
iex> t = Nx.tensor([[1, 2], [3, 4], [5, 6], [7, 8]])
iex> t[0..1]
#Nx.Tensor<
  s32[2][2]
  [
    [1, 2],
    [3, 4]
  ]
>
Ranges can receive negative positions and they will read from
the back. In such cases, the range step must be explicitly given
and the right-side of the range must be equal or greater than
the left-side:
iex> t = Nx.tensor([[1, 2], [3, 4], [5, 6], [7, 8]])
iex> t[1..-2//1]
#Nx.Tensor<
  s32[2][2]
  [
    [3, 4],
    [5, 6]
  ]
>
As you can see, accessing with a range does not eliminate the
accessed axis. This means that, if you try to cascade ranges,
you will always be filtering the highest dimension:
iex> t = Nx.tensor([[1, 2], [3, 4], [5, 6], [7, 8]])
iex> t[1..-1//1] # Drop the first "row"
#Nx.Tensor<
  s32[3][2]
  [
    [3, 4],
    [5, 6],
    [7, 8]
  ]
>
iex> t[1..-1//1][1..-1//1] # Drop the first "row" twice
#Nx.Tensor<
  s32[2][2]
  [
    [5, 6],
    [7, 8]
  ]
>
Therefore, if you want to slice across multiple dimensions, you can wrap
the ranges in a list:
iex> t = Nx.tensor([[1, 2], [3, 4], [5, 6], [7, 8]])
iex> t[[1..-1//1, 1..-1//1]] # Drop the first "row" and the first "column"
#Nx.Tensor<
  s32[3][1]
  [
    [4],
    [6],
    [8]
  ]
>
You can also use .. as the full-slice range, which means you want to
keep a given dimension as is:
iex> t = Nx.tensor([[1, 2], [3, 4], [5, 6], [7, 8]])
iex> t[[.., 1..-1//1]] # Drop only the first "column"
#Nx.Tensor<
  s32[4][1]
  [
    [2],
    [4],
    [6],
    [8]
  ]
>
You can mix both ranges and integers in the list too:
iex> t = Nx.tensor([[1, 2, 3], [4, 5, 6], [7, 8, 9], [10, 11, 12]])
iex> t[[1..2, 2]]
#Nx.Tensor<
  s32[2]
  [6, 9]
>
If the list has less elements than axes, the remaining dimensions
are returned in full:
iex> t = Nx.tensor([[1, 2, 3], [4, 5, 6], [7, 8, 9], [10, 11, 12]])
iex> t[[1..2]]
#Nx.Tensor<
  s32[2][3]
  [
    [4, 5, 6],
    [7, 8, 9]
  ]
>
The access syntax also pairs nicely with named tensors. By using named
tensors, you can pass only the axis you want to slice, leaving the other
axes intact:
iex> t = Nx.tensor([[1, 2, 3], [4, 5, 6], [7, 8, 9], [10, 11, 12]], names: [:x, :y])
iex> t[x: 1..2]
#Nx.Tensor<
  s32[x: 2][y: 3]
  [
    [4, 5, 6],
    [7, 8, 9]
  ]
>
iex> t[x: 1..2, y: 0..1]
#Nx.Tensor<
  s32[x: 2][y: 2]
  [
    [4, 5],
    [7, 8]
  ]
>
iex> t[x: 1, y: 0..1]
#Nx.Tensor<
  s32[y: 2]
  [4, 5]
>
For a more complex slicing rules, including strides, you
can always fallback to Nx.slice/4.

  
    
  
  Backends


The Nx library has built-in support for multiple backends.
A tensor is always handled by a backend, the default backend
being Nx.BinaryBackend, which means the tensor is allocated
as a binary within the Erlang VM.
Most often backends are used to provide a completely different
implementation of tensor operations, often accelerated to the GPU.
In such cases, you want to guarantee all tensors are allocated in
the new backend. This can be done by configuring your runtime:
# config/runtime.exs
import Config
config :nx, default_backend: EXLA.Backend
In your notebooks and on Mix.install/2, you might:
Mix.install(
  [
    {:nx, ">= 0.0.0"}
  ],
  config: [nx: [default_backend: EXLA.Backend]]
)
Or by calling Nx.global_default_backend/1 (less preferable):
Nx.global_default_backend(EXLA.Backend)
To pass options to the backend, replacing EXLA.Backend by
{EXLA.Backend, client: :cuda} or similar. See the documentation
for EXLA and Torchx
for installation and GPU support.
To implement your own backend, check the Nx.Tensor behaviour.

Common constants used in computations.
This module can be used in defn.
The functions e/0, pi/0 and i/0 will follow the same rules as
literal constants when used inside defn. This means that they will
use the surrounding precision instead of defaulting to f32.

Numerical functions.
A numerical function is a subset of Elixir tailored for
numerical computations. For example, the following function:
defmodule MyModule do
  import Nx.Defn

  defn softmax(t) do
    Nx.exp(t) / Nx.sum(Nx.exp(t))
  end
end
will work with scalars, vector, matrices, and n-dimensional
tensors. Depending on your compiler of choice, the code can even
be JIT-compiled and run either on the CPU or GPU.
To support these features, defn is a subset of Elixir. It
replaces Elixir's Kernel by Nx.Defn.Kernel. Nx.Defn.Kernel
provides tensor-aware operators, such as +, -, etc, while
also preserving many high-level constructs known to Elixir
developers, such as pipe operator, aliases, conditionals,
pattern-matching, the access syntax, and more:
For example, the code above can also be written as:
defmodule MyModule do
  import Nx.Defn

  defn softmax(t) do
    t
    |> Nx.exp()
    |> then(& &1 / Nx.sum(&1))
  end
end
Please consult Nx.Defn.Kernel for a complete reference.
Some of the functions in this module may also be used within
defn.

  
    
  
  Operators


defn attempts to keep as close to the Elixir semantics as
possible but that's not achievable. For example, mathematical
and bitwise operators (+, -, &&&, <<<, etc.) in Elixir
work on numbers, which means mapping them to tensors is
straight-forward and they largely preserve the same semantics,
except they are now multi-dimensional.
On the other hand, the logical operators and, or, and not
work with booleans in Elixir (true and false), which map
to 0 and 1 in defn.
Therefore, when working with logical operators inside defn,
0 is considered false and all other numbers are considered
true, which is represented as the number 1. For example, in
defn, 0 and 1 as well as 0 and 2 return 0, while
1 and 1 or 1 and -1 will return 1.
The same semantics apply to conditional expressions inside defn,
such as if, while, etc.

  
    
  
  JIT compilers


The power of Nx.Defn is given by its compilers. The default
compiler is Nx.Defn.Evaluator, which evaluates the code.
You can use jit/3 to compile a function on the fly using a
different compiler, such as EXLA:
fun = Nx.Defn.jit(&MyModule.softmax/1, compiler: EXLA)
fun.(my_tensor)
The above will return an anonymous function that optimizes,
compiles, and run softmax on the fly on the CPU (or the GPU)
if available. EXLA, in particular, also exports a EXLA.jit/2
function for convenience.
defn functions are compiled when they are invoked, based on
the type and shapes of the tensors given as arguments.
Therefore compilation may be quite time consuming on the first
invocation. The compilation is then cached based on the tensors
shapes and types. Calling the same function with a tensor of
different values but same shape and type means no recompilation
is performed.
For those interested in writing custom compilers, see Nx.Defn.Compiler.

  
    
  
  Invoking custom Elixir code


Inside defn you can only call other defn functions and
the functions in the Nx module. However, it is possible
to use transforms, defined with either deftransform or
deftransformp to invoke any Elixir code.
You can call code which was defined with deftransform from another module:
defmodule MyRemoteModule do
  import Nx.Defn

  deftransform remote_elixir_code(value) do
    IO.inspect(value)
  end
end

defn add_and_mult(a, b, c) do
  res = a * b + c
  MyRemoteModule.remote_elixir_code(res)
end
You can also define and call a private transform defined through deftransformp:
defn add_and_mult(a, b, c) do
  res = a * b + c
  custom_elixir_code(res)
end

deftransformp custom_elixir_code(value), do: IO.inspect(value)
The only difference between using deftransform and deftransformp
is whether you want to expose and share the code with other modules,
just like def and defp.
Transforms are useful to manipulate tensor expressions or
Elixir data structures without the constraints of defn.

  
    
  
  Inputs and outputs types


Nx and defn expect the arguments to be numbers, tensors,
or one composite data type that implements Nx.LazyContainer.
Tuples and maps implement Nx.LazyContainer by default.
As previously described, defn are cached based on the shape,
type, and names of the input tensors, but not their values.
defn also accepts two special arguments: functions (or tuples
of functions) and lists (most commonly as keyword lists). Those
values are passed as is to numerical definitions and cached as
a whole. For this reason, you must never capture tensors in
functions or pass tensors in keyword lists.
When numbers are given as arguments, they are always immediately
converted to tensors on invocation. If you want to keep numbers
as is or if you want to pass any other value to numerical definitions,
they must be given as keyword lists.

  
    
  
  Default arguments


defn functions support default arguments. They are typically used
as options. For example, imagine you want to create a function named
zeros, which returns a tensor of zeroes with a given type and shape.
It could be implemented like this:
defn zeros(opts \\ []) do
  opts = keyword!(opts, type: {:f, 32}, shape: {})
  Nx.broadcast(Nx.tensor(0, type: opts[:type]), opts[:shape])
end
The function above accepts opts which are then validated and given
default values via the keyword!/2 function. Note that while it is
possible to access options via the Access syntax, such as opts[:shape],
it is not possible to directly call functions in the Keyword module
inside defn. To freely manipulate any Elixir value inside defn,
you have to use transforms, as described in the "Invoking custom Elixir
code" section.
Important! When it comes to JIT compilation, each different set of
options (as well as anonymous functions) will lead to a different
compilation of the numerical function.
Furthermore, if tensors are given through keyword lists, they won't
be cached effectively. Tensors in defn are cached based on their shape
and type, not their value, but this is not true if the tensor is given
via a default argument or captured by an anonymous function. For this
reason, it is extremely discouraged to pass tensors through anonymous
functions and default arguments.


  
    
  
  Working with maps and structs


While Nx supports maps in defn, you must be careful if your numerical
definitions are receiving maps and returning maps. For example, imagine
this code:
defn update_a(map) do
  %{map | a: Nx.add(map.a, 1)}
end
The following code increments the value under the key :a
by 1. However, because the function receives the whole map and
returns the whole map, it means if the map has 120 keys, the
whole map will be copied to the CPU/GPU, and then brought back.
However, if you do this instead:
defn update_a(map) do
  Nx.add(map.a, 1)
end
And then update the map on Elixir, outside of defn:
%{map | a: update_a(map)}
Nx will only send the parts of the map that matters.

  
    
  
  Recursion and loops


Given numerical definition first build a representation of
your code, it is not possible to write recursive (nor tail
recursive) code inside defn. Instead, one must use
Nx.Defn.Kernel.while/4.

A module for splitting Nx.Defn.Expr into stages.
This module is used to split an Nx.Defn.Expr into stages, which are then
executed in a chain.
split/2 and t:Stage.t() describe how to split
the graph and what's the expected result.
run/2 executes the given graph against the provided arguments in a sequential manner.

All imported functionality available inside defn blocks.
This module can be used in defn.

Nx conveniences for linear algebra.
This module can be used in defn.

Parallel Jacobi symmetric eigendecomposition.
Reference implementation taking from XLA's eigh_expander
which is built on the approach in:
Brent, R. P., & Luk, F. T. (1985). The solution of singular-value
and symmetric eigenvalue problems on multiprocessor arrays.
SIAM Journal on Computing, 6(1), 69-84. https://doi.org/10.1137/0906007

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Nx.LinAlg.Cholesky - Nx v0.10.0
    
    

    


  
  

    
Nx.LinAlg.Cholesky 
    




      
        Summary


  
    Functions
  


    
      
        
  
    
    Nx.LinAlg.Eigh - Nx v0.10.0
    
    

    


  
  

    
Nx.LinAlg.Eigh 
    




      
        Summary


  
    Functions
  


    
      
        
  
    
    Nx.LinAlg.QR - Nx v0.10.0
    
    

    


  
  

    
Nx.LinAlg.QR 
    




      
        Summary


  
    Functions
  


    
      
        
  
    
    Nx.Pointer - Nx v0.10.0
    
    

    


  
  

    
Nx.Pointer 
    



      
Represents a reference to a value in memory.
Can represent either a pointer or an IPC handle.

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Nx.Random - Nx v0.10.0
    
    

    


  
  

    
Nx.Random 
    



      
Pseudo-random number generators.
Unlike the stateful pseudo-random number generators (PRNGs)
that users of most programming languages and numerical libraries
may be accustomed to, Nx random functions require an explicit
PRNG key to be passed as a first argument (see below for more info). That key is defined by
an Nx.Tensor composed of 2 unsigned 32-bit integers, usually
generated by the Nx.Random.key/1 function:
iex> Nx.Random.key(12)
#Nx.Tensor<
  u32[2]
  [0, 12]
>
Or for example:
iex> Nx.Random.key(System.os_time())
This key can then be used in any of Nx’s random number generation
routines:
iex> key = Nx.Random.key(12)
iex> {uniform, _new_key} = Nx.Random.uniform(key)
iex> uniform
#Nx.Tensor<
  f32
  0.7691127061843872
>
Now, when generating a new random number, you pass the new_key
to get a different number.
The function in this module also have a *_split variant, which
is used when the key has been split before hand.

  
    
  
  Design and Context


In short, Nx's PRNGs are based on a Threefry counter PRNG
associated to a functional array-oriented splitting model.
To summarize, among other requirements, Nx's PRNG aims to:
	Ensure reproducibility

	Parallelize well, both in terms of vectorization
(generating array values) and multi-replica, multi-core
computation. In particular it should not use sequencing
constraints between random function calls.



  
    
  
  The key to understanding Nx.Random keys


Most Elixir users might be used to not having to keep
track of the PRNG state while their code executes.
While this works fine when we're dealing with the CPU,
we can think of keeping track of the Nx.Random key as a way to
isolate multiple GPU users, much like the PRNG on different
BEAM nodes is isolated. Each key gets updated in its own
isolated sequence of calls, and thus we don't get different
results for each process using the same PRNG as we would
in the normal situation.
The fact that the key is a parameter for the functions also
helps with the caching and operator fusion of the computational
graphs. Because the PRNG functions themselves are stateless,
compilers can take advantage of this to further improve execution times.

Serving encapsulates client and server work to perform batched requests.
Servings can be executed on the fly, without starting a server, but most
often they are used to run servers that batch requests until a given size
or timeout is reached.
More specifically, servings are a mechanism to apply a computation on a
Nx.Batch, with hooks for preprocessing input from and postprocessing
output for the client. Thus we can think of an instance of Nx.Serving.t/0
(a serving) as something that encapsulates batches of Nx computations.

  
    
  
  Inline/serverless workflow


First, let's define a simple numerical definition function:
defmodule MyDefn do
  import Nx.Defn

  defn print_and_multiply(x) do
    x = print_value(x, label: "debug")
    x * 2
  end
end
The function prints the given tensor and doubles its contents.
We can use new/1 to create a serving that will return a JIT
or AOT compiled function to execute on batches of tensors:
iex> serving = Nx.Serving.new(fn opts -> Nx.Defn.jit(&MyDefn.print_and_multiply/1, opts) end)
iex> batch = Nx.Batch.stack([Nx.tensor([1, 2, 3])])
iex> Nx.Serving.run(serving, batch)
debug: #Nx.Tensor<
  s64[1][3]
  [
    [1, 2, 3]
  ]
>
#Nx.Tensor<
  s64[1][3]
  [
    [2, 4, 6]
  ]
>
We started the serving by passing a function that receives
compiler options and returns a JIT or AOT compiled function.
We called Nx.Defn.jit/2 passing the options received as
argument, which will customize the JIT/AOT compilation.
You should see two values printed. The former is the result of
Nx.Defn.Kernel.print_value/1, which shows the tensor that was
actually part of the computation and how it was batched.
The latter is the result of the computation.
When defining a Nx.Serving, we can also customize how the data is
batched by using the client_preprocessing as well as the result by
using client_postprocessing hooks. Let's give it another try,
this time using jit/2 to create the serving, which automatically
wraps the given function in Nx.Defn.jit/2 for us:
iex> serving = (
...>   Nx.Serving.jit(&MyDefn.print_and_multiply/1)
...>   |> Nx.Serving.client_preprocessing(fn input -> {Nx.Batch.stack(input), :client_info} end)
...>   |> Nx.Serving.client_postprocessing(&{&1, &2})
...> )
iex> Nx.Serving.run(serving, [Nx.tensor([1, 2]), Nx.tensor([3, 4])])
debug: #Nx.Tensor<
  s64[2][2]
  [
    [1, 2],
    [3, 4]
  ]
>
{{#Nx.Tensor<
    s64[2][2]
    [
      [2, 4],
      [6, 8]
    ]
  >,
  :server_info},
 :client_info}
You can see the results are a bit different now. First of all, notice that
we were able to run the serving passing a list of tensors. Our custom
client_preprocessing function stacks those tensors into a batch of two
entries and returns a tuple with a Nx.Batch struct and additional client
information which we represent as the atom :client_info. The default
client preprocessing simply enforces a batch (or a stream of batches)
was given and returns no client information.
Then the result is a triplet tuple, returned by the client
postprocessing function, containing the result, the server information
(which we will later learn how to customize), and the client information.
From this, we can infer the default implementation of client_postprocessing
simply returns the result, discarding the server and client information.
So far, Nx.Serving has not given us much. It has simply encapsulated the
execution of a function. Its full power comes when we start running our own
Nx.Serving process. That's when we will also learn why we have a client_
prefix in some of the function names.

  
    
  
  Stateful/process workflow


Nx.Serving allows us to define an Elixir process to handle requests.
This process provides several features, such as batching up to a given
size or time, partitioning, and distribution over a group of nodes.
To do so, we need to start a Nx.Serving process with a serving inside
a supervision tree:
children = [
  {Nx.Serving,
   serving: Nx.Serving.jit(&MyDefn.print_and_multiply/1),
   name: MyServing,
   batch_size: 10,
   batch_timeout: 100}
]

Supervisor.start_child(children, strategy: :one_for_one)
Note: in your actual application, you want to make sure
Nx.Serving comes early in your supervision tree, for example
before your web application endpoint or your data processing
pipelines, as those processes may end-up hitting Nx.Serving.

Now you can send batched runs to said process:
iex> batch = Nx.Batch.stack([Nx.tensor([1, 2, 3]), Nx.tensor([4, 5, 6])])
iex> Nx.Serving.batched_run(MyServing, batch)
debug: #Nx.Tensor<
  s64[2][3]
  [
    [1, 2, 3],
    [4, 5, 6]
  ]
>
#Nx.Tensor<
  s64[2][3]
  [
    [2, 4, 6],
    [8, 10, 12]
  ]
>
In the example, we pushed a batch of 2 and eventually got a reply.
The process will wait for requests from other processes, for up to
100 milliseconds or until it gets 10 entries. Then it merges all
batches together and once the result is computed, it slices and
distributes those responses to each caller.
If there is any client_preprocessing function, it will be executed
before the batch is sent to the server. If there is any client_postprocessing
function, it will be executed after getting the response from the
server.

  
    
  
  Partitioning


You can start several partitions under the same serving by passing
partitions: true when starting the serving. The number of partitions
will be determined according your compiler and for which host it is
compiling.
For example, when creating the serving, you may pass the following
defn_options:
Nx.Serving.new(computation, compiler: EXLA, client: :cuda)
Now when booting up the serving:
children = [
  {Nx.Serving,
   serving: serving,
   name: MyServing,
   batch_size: 10,
   batch_timeout: 100,
   partitions: true}
]
If you have two GPUs, batched_run/3 will now gather batches and send
them to the GPUs as they become available to process requests.
Cross-device operations
When partitions: true is set, you will receive results from
different GPU devices and Nx won't automatically transfer data
across devices to avoid surprising performance pitfalls, which
may lead to errors. In such cases, you probably want to transfer
tensors back to host on your serving execution.


  
    
  
  Distribution


All Nx.Servings are distributed by default. If the current machine
does not have an instance of Nx.Serving running, batched_run/3 will
automatically look for one in the cluster. The nodes do not need to run
the same code and applications. It is only required that they run the
same Nx version.
The load balancing between servings is done randomly by default, however,
the number of partitions are considered if the partitions: true option is also given.
For example, if you have a node with 2 GPUs and another with 4, the latter
will receive the double of requests compared to the former.
Furthermore, the load balancing allows for assigning weights to servings.
Similarly to the number of partitions, when running a serving with distribution_weight: 1
and another one with distribution_weight: 2, the latter will receive double the requests
compared to the former.
batched_run/3 receives an optional distributed_preprocessing callback as
third argument for preprocessing the input for distributed requests. When
using libraries like EXLA or Torchx, the tensor is often allocated in memory
inside a third-party library so it is necessary to either transfer or copy
the tensor to the binary backend before sending it to another node.
This can be done by passing either Nx.backend_transfer/1 or Nx.backend_copy/1
as third argument:
Nx.Serving.batched_run(MyDistributedServing, input, &Nx.backend_copy(&1, Nx.BinaryBackend))
Use backend_transfer/1 if you know the input will no longer be used.
Similarly, the serving has a distributed_postprocessing callback which is
called on the remote machine before sending the reply to the caller. It can
be used to transfer resources to the binary backend before sending them over
the network.
The servings are dispatched using Erlang Distribution. You can use
Node.connect/1 to manually connect nodes. In a production setup, this is
often done with the help of libraries like libcluster.

  
    
  
  Advanced notes

